60496900

CONTROL DATA CORPORATION

COBOL VERSION 5 REPORT WRITER USER'S GUIDE

CDC® OPERATING SYSTEMS:
NOS 1
NOS/BE 1

REVISION RECORD	
REVISION	**DESCRIPTION**
A	Original release, PSR level 439.
(11-15-76)	
Publication No. 60496900	

REVISION LETTERS I, O, Q AND X ARE NOT USED

Address comments concerning this manual to:

CONTROL DATA CORPORATION
Publications and Graphics Division
215 MOFFETT PARK DRIVE
SUNNYVALE, CALIFORNIA 94086

or use Comment Sheet in the back of this manual

LIST OF EFFECTIVE PAGES

ew features, as well as changes, deletions, and additions to information in this manual are indicated by bars in the argins or by a dot near the page number if the entire page is affected. A bar by the page number indicates pagina-on rather than content has changed.

Page	Revision
Cover	—
Title Page	—
ii	A
iii/iv	A
v/vi	A
vii, viii	A
1-1	A
2-1, 2-2	A
3-1 thru 3-9	A
4-1 thru 4-18	A
5-1 thru 5-8	A
6-1 thru 6-6	A
A-1 thru A-8	A
B-1 thru B-4	A
C-1 thru C-7	A
Index-1, -2	A
Comment Sht.	A
Reply Env.	—
Cover	—

Page	Revision

Page	Revision

PREFACE

This guide describes the usage of the Report Writer feature of the COBOL Version 5.1 language. COBOL 5 is implemented for the CONTROL DATA® CYBER 170 Series, CYBER 70 Models 71, 72, 73, and 74, and 6000 Series Computer Systems and operates under control of the NOS 1 and NOS/BE 1 operating systems.

The Report Writer feature of COBOL 5 is designed as specified in American National Standard X3.23-1974, COBOL.

This guide is written for a programmer familiar with the COBOL 5 language and with the operating system under which the COBOL 5 compiler is operating. Only those clauses and statements that are specifically related to the Report Writer feature are described in this guide.

Detailed information can be found in the publications listed below.

Publication	Publication Number
NOS 1 Reference Manual, Volume 1	60435400
NOS 1 Reference Manual, Volume 2	60435300
NOS/BE 1 Reference Manual	60493800
COBOL Version 5 Reference Manual	60497100
COBOL Version 5 User's Guide	60497200
COBOL Version 4 to COBOL Version 5 Conversion Aid Reference Manual	19265021
FORM Version 1 Reference Manual	60496200

CDC® manuals can be ordered from Control Data Literature and Distribution Services, mailing address Box 0, Minneapolis, Minnesota, 55440.

This product is intended for use only as described in this document. Control Data cannot be responsible for the proper functioning of undescribed features or parameters.

CONTENTS

APPENDIXES

FIGURES

TABLES

INTRODUCTION TO REPORT WRITER 1

The Report Writer feature of the COBOL 5 language provides a concise means for structuring and generating printed reports. When this feature is used, the programmer is relieved of writing the procedures to produce the printed pages. Report Writer performs these operations automatically.

Only three Procedure Division statements are required to generate the report. The INITIATE statement initializes the counters and the special registers for the report. The GENERATE statement causes the report lines to be written on the report file. The TERMINATE statement performs end-of-report processing. The programmer does not have to be concerned with such operations as moving data into the print lines, counting the lines on a page and the pages in the report, producing heading and footing lines, and accumulating subtotals.

The structure of the report is described in the Report Section of the Data Division. Each different type of print line is called a report group and is described by a Report Group Description entry, which is similar to a Record Description entry in the File Section of the Data Division. The types of report groups that can be described are a report heading, a page heading, control headings, detail lines, control footings, a page footing, and a report footing. A report group can include more than one print line. The clauses included in the individual entries within a Report Group Description entry specify the data to be printed and the positioning of the data on the page.

The data to be printed on a line can be a data item described in the Data Division, a constant value defined by the VALUE clause, or a total accumulated by Report Writer. Data items and constant values can be specified for any type of report group. Totals accumulated by Report Writer can only be specified for control footing report groups.

Automatic totaling is performed by Report Writer when the SUM clause is specified for a printable item. This clause establishes a sum counter and specifies the items to be summed. Three types of summing are available: subtotaling, crossfooting, and rolling forward. Subtotaling is performed when a data item is added to the sum counter. Crossfooting is adding a sum counter to another sum counter in the same control footing report group. Rolling forward is adding a sum counter to another sum counter in a higher level control footing report group. Subtotaling occurs whenever a detail line is generated. Crossfooting and rolling forward are performed during the processing of control footing report groups.

Control heading and control footing report groups are written on the report file as the result of a change in the value of a specified control data item. When the value changes, a control break occurs and Report Writer automatically generates the applicable control footings and control headings before the detail line is written on the report file.

The number of control breaks monitored by Report Writer and the hierarchy of the control breaks are established by the CONTROL clause. The control data items to be monitored are specified from high level (major control break) to low level (minor control break). A control heading and/or a control footing report group is associated with each break control item. When a control break occurs, the control report groups written on the report file are those associated with the item that caused the control break and with all lower level break control items.

More than one report can be generated by Report Writer during program execution. When multiple reports are written, print lines for all the reports are written on the report file in the order in which the lines are generated. At program termination, the print lines are intermixed on the report file and need to be separated by a user program before the reports can be printed. A code used during report generation identifies the print lines for a specific report.

Declarative procedures can be executed during report generation. The USE BEFORE REPORTING statement in the Declaratives portion of the Procedure Division specifies the data-name of a report group. During program execution, the declarative procedure is executed immediately before the specified report group is processed.

As the program executes, report groups are automatically written on the report file. If the report file is assigned to the system file OUTPUT, the report is printed at job termination. The report file is output to the printer at the central site or remote terminal at which the job was submitted. The ROUTE control statement can be used to output the report file to a terminal or site other than the originating one. Refer to the operating system reference manual for a description of the ROUTE control statement. If the report is not printed at job termination, the report file should be preserved for later processing.

The major advantage of Report Writer is the ability to specify a hierarchy of control breaks according to a predefined pattern. Automatic processing of control breaks, including the summing of specified items, relieves the programmer of writing the Procedure Division paragraphs necessary to monitor the control breaks and to produce the various totals. In some cases, however, the requirements for a report cannot be met easily by Report Writer usage. For example, Report Writer cannot generate crosstotals of items in a detail line. Other report requirements that cannot be met by Report Writer are to reset a sum counter to a nonzero value and to print control headings after a page eject. Report specifications should be analyzed to determine whether or not it is advantageous to use Report Writer.

Section 2 discusses briefly the Report Writer elements of a COBOL 5 source program to provide a general description of these elements and to establish the Report Writer terminology used in the guide. Detailed discussions and sample programs are presented in section 3 through section 6. Language formats applicable to Report Writer are summarized in appendix B; complete reference information can be found in the COBOL 5 Reference Manual. For comparison purposes, appendix C shows a COBOL 5 program that generates the report shown in figure 4-14 without using Report Writer.

Certain elements in a COBOL 5 source program are specifically related to automatic report generation through the Report Writer feature. The report file is named in the File Section; however, the description of the report file is specified in the Report Section. Two special registers are provided for automatic line counting and page counting. The report file is generated when Report Writer statements in the Procedure Division are executed.

FILE DESCRIPTION ENTRY

The File Description entry for a report file specifies the program file-name of the report file and the report-name of each report to be written on the file. The report file-name must be the file-name of a sequential file specified in a SELECT clause in the Environment Division. Only two clauses are applicable to the File Description entry for a report file.

The REPORT clause specifies the name of each report to be written on the report file. When more than one report-name is specified, multiple reports are created; refer to section 5 for the discussion on generating multiple reports. The report named in this clause is described by a Report Description entry in the Report Section. No Record Description entries can be included in the File Description entry for a report file.

The LABEL RECORDS clause, which must be included in every File Description entry, specifies OMITTED or STANDARD. The LABEL RECORDS ARE STANDARD clause can only be used when the report file is written on magnetic tape.

Records written on a report file are either 137 or 139 characters in length. When a single report is being created, 137-character records are written on the report file. A carriage control character is the first character in the record; this character is supplied by the system. The remaining 136 characters correspond to the 136 character positions in a print line and contain data formatted according to the Report Description entry. If the report file is assigned to the system file OUTPUT, the report is printed at job termination; otherwise, the report file must be preserved by the job for later processing. When multiple reports are being created, 139-character records are written on the report file. A two-character code is followed by the carriage control character and the 136 characters of the print line (refer to section 5).

A File Description entry for a report file is illustrated in figure 2-1. The report file-name is OUT-FILE; the file is assigned to the system file OUTPUT so that the report will be printed at job termination. The REPORT clause specifies that the report-name is SALES.

SPECIAL REGISTERS

Report Writer automatically creates and maintains two special registers for each report being generated. Both special registers can be referenced within the Report Description entry as source items for a print line.

```
ENVIRONMENT DIVISION.
      .
      .
      SELECT OUT-FILE ASSIGN TO "OUTPUT".
      .
      .
DATA DIVISION.
FILE SECTION.
FD    OUT-FILE
            LABEL RECORDS ARE OMITTED
            REPORT IS SALES.
```

Figure 2-1. File Description Entry for a Report File

PAGE-COUNTER

The special register PAGE-COUNTER is used to number the pages of the report being generated. The register is set to 1 when the report is initiated and is incremented by 1 each time the report is advanced to the next page.

PAGE-COUNTER is a 6-digit COMPUTATIONAL-1 integer; it is unsigned. The value of PAGE-COUNTER can be used in a page heading or page footing to print page numbers on each page of the report. PAGE-COUNTER can be referenced in Procedure Division statements as well as in the Report Description entry. The value of the register can be changed by Procedure Division statements. If more than one report is being generated, PAGE-COUNTER must be qualified by the report-name in Procedure Division references; however, it does not have to be qualified within the Report Section.

LINE-COUNTER

The line number on which a print line is written is controlled by the special register LINE-COUNTER. This register is set to zero when the report is initiated. As report lines are being generated, LINE-COUNTER is incremented according to the line specification for each type of print line processed. Before a line is written on the report file, LINE-COUNTER is incremented as specified in the LINE NUMBER clause. After the complete report group has been processed, LINE-COUNTER is incremented according to the NEXT GROUP clause, if specified. When a page advance occurs, LINE-COUNTER is reset to zero.

Report Writer maintains LINE-COUNTER as an unsigned 6-digit COMPUTATIONAL-1 integer item. When a line is being processed, the value of LINE-COUNTER indicates the line number on which the line is to be printed. The register can be specified as a source item for a print line. It can also be referenced in Procedure Division statements; however, the value of LINE-COUNTER cannot be changed by the statements. If more than one report is being created, Procedure Division references to LINE-COUNTER must be qualified by the report-name; references within the Report Section need not be qualified.

REPORT SECTION

The detailed description of the report is specified in the Report Section. Each report named in the File Section is described by a Report Description entry, which consists of an RD entry followed by Report Group Description entries. The Report Section must be the last section in the Data Division.

REPORT DESCRIPTION ENTRY

A Report Description entry contains information required by Report Writer in order to generate a report. The complete entry describes in detail the structure and organization of the report. It includes the format of a page and the format of each type of line that can be printed in the report.

The RD entry specifies the report-name; this must be a report-name specified in a File Description entry in the File Section. Optional clauses are included in the RD entry to define the page limits and the areas in which specific types of lines can be printed and to specify the break control items for control headings and control footings. When multiple reports are generated, a code can be specified for identifying the lines belonging to a specific report.

REPORT GROUP DESCRIPTION ENTRY

A Report Group Description entry specifies the characteristics of a report group to be presented in the report. A report group is generally one line on a page; however, it can include more than one line. At least one Report Group Description entry must follow the RD entry.

Seven types of report groups can be specified: three different headings, detail lines, and three different footings. A heading report group and a footing report group can be specified for the report, for each page, and for control breaks. Control headings, detail lines, and control footings are considered body report groups; at least one body group must be specified for the report.

Each Report Group Description entry begins with a level 01 entry, which is followed by any number of higher level entries. The level 01 entry identifies the type of report group being described. The higher level entries are subordinate group and elementary entries that specify the positioning and the source of the items in the report group.

REPORT WRITER STATEMENTS

The report is generated and written on the report file when Report Writer statements in the Procedure Division are executed. The only other statements that can reference the report file are the OPEN and CLOSE statements. The report file must be opened with the OPEN OUTPUT or OPEN EXTEND statement before any Report Writer statement is executed. The CLOSE statement is used to close the report file after all Report Writer statements have been executed.

INITIATE STATEMENT

The INITIATE statement must be the first Report Writer statement that is executed. This statement specifies the report-name of each report being generated. When the INITIATE statement is executed, all sum counters and the special register LINE-COUNTER are set to zero; the special register PAGE-COUNTER is set to 1.

GENERATE STATEMENT

Report Writer produces the report as a result of the execution of a GENERATE statement. This statement specifies that either a summary report or a detail report group is written on the report file. Control breaks are also processed during execution of a GENERATE statement.

Detail Report Group Processing

When the GENERATE statement specifies the data-name of a detail report group, the report group is processed and written on the report file. The first execution of a GENERATE statement causes Report Writer to save the values of any control items defined for the report. Execution of subsequent GENERATE statements causes the control items to be checked for a control break. If a control break is encountered, the detail report group is processed after the applicable control report groups are processed.

Before the first detail report group is processed, all report heading, page heading, and control heading report groups specified for the report are processed and written on the report file. As control breaks and page advances occur during subsequent execution of GENERATE statements, the appropriate heading and footing report groups are processed. Control report groups and their presentation in the report are discussed in detail in section 4.

Summary Report Processing

Report Writer produces a summary report when the GENERATE statement specifies the report-name. Detail report groups are not written on the report file when a summary report is generated.

A summary report can be specified only if the Report Description entry for the report satisfies certain requirements. The RD entry must include the CONTROL clause to specify break control items. At least one body group (detail, control heading, or control footing report group) must be included; no more than one detail report group can be specified.

When a GENERATE statement is executed for a summary report, the detail report group is processed but is not written on the report file. Any specified subtotaling of items in the detail report group is performed.

TERMINATE STATEMENT

The TERMINATE statement is the last Report Writer statement that is executed. Each report that is initialized by the INITIATE statement is specified in a TERMINATE statement. The report file is not closed by this statement; a CLOSE statement must be executed after the TERMINATE statement to close the report file.

When the TERMINATE statement is executed, termination processing completes the report by presenting the applicable footing report groups. All specified control footing report groups are processed first followed by the page footing and report footing report groups. If a page advance is required during termination processing, page footing and page heading report groups are processed in conjunction with the page advance.

SUPPRESS STATEMENT

The SUPPRESS statement is used to suppress the printing of a report group. It can only be specified in the Declaratives portion of the Procedure Division. The SUPPRESS statement is discussed in detail in section 6.

Report Writer generates the report according to the specifications of the Report Description entry in the Report Section of the source program. Before the report groups can be described, the format and content of each line to be printed in the report must be determined. This section of the guide presents the step-by-step procedure for describing a report that is simply a reformatted listing of the input data. Control breaks in a report are discussed in section 4.

DETERMINING THE REPORT FORMAT

The first step in creating any report is determining the structure and organization of the printed page. Each line to be printed is formatted to show vertical line spacing and horizontal positioning of the printable items.

A report worksheet that indicates column numbers and line numbers is helpful in formatting the report. It shows the physical layout of a page and provides the column numbers and line numbers needed to describe the report in the Report Description entry. The report worksheet for the report described in this section is shown in figure 3-1.

SPECIFYING THE PAGE LIMITS

The length of a report page is defined by the PAGE clause in the RD entry. Optional phrases are included in this clause to subdivide the page into areas within which the various report groups can appear. If the PAGE clause is not specified, the report is a single page of indefinite length.

The PAGE clause specifies the number of lines that are available on each page. If none of the optional phrases are included in this clause, report groups are printed beginning with line number 1 and ending with the specified line number. A page advance occurs when printing of a report group would extend beyond the limit specified in the PAGE clause.

The optional phrases in the PAGE clause are used to divide the report page into three areas. The heading area consists of the lines on which report heading and page heading report groups can be printed. The body area consists of the lines on which body report groups (detail and control report groups) can be printed. Report footing and page footing report groups can be printed on the lines designated as the footing area.

The HEADING phrase specifies the line number of the first line in the heading area. If the phrase is omitted, line number 1 is assumed by default. The Report Group Description entry for a report heading or a page heading cannot specify a line number less than the first line of the heading area. The heading area extends to the line immediately preceding the first line of the body area.

The FIRST DETAIL phrase specifies the line number of the first line in the body area. The line number in this phrase cannot be less than the line number specified for the heading area. If the FIRST DETAIL phrase is not included in the PAGE clause, the first line of the heading area is assumed to be the first line of the body area.

The body area extends to the line specified in the LAST DETAIL phrase. If this phrase is not specified, the area extends to the first line of the footing area or to the page limit if no footing area is defined for the page.

The FOOTING phrase applies to control footing report groups and is described with respect to control breaks in section 4. If the phrase is specified for a report that has no control footings, the footing area begins with the line following the line number specified in the FOOTING phrase; otherwise, the footing area begins with the line following the line number specified in the LAST DETAIL phrase. The footing area extends to the line defined as the page limit. If neither the LAST DETAIL phrase nor the FOOTING phrase is included in the PAGE clause, no footing area is defined for the page. Report footing and page footing report groups must appear within the footing area.

Table 3-1 summarizes the limits of the heading, body, and footing areas.

The RD entry for the SALES report is shown in figure 3-2. The page is defined as having a length of 50 lines. The heading area consists of lines 3 through 9, the body area consists of lines 10 through 40, and the footing area consists of lines 41 through 50.

DESCRIBING THE REPORT GROUPS

Each type of line to be generated for the report is described by a Report Group Description entry. These entries immediately follow the RD entry for the report.

A Report Group Description entry consists of a level 01 entry followed by at least one higher level entry. The level 01 entry identifies the type of report group being defined by the TYPE clause. This entry can also include the LINE NUMBER clause when the report group is presented on a single line; the NEXT GROUP clause is included to specify line spacing before the next report group is presented. The subordinate entries are group and elementary items that specify the data to be printed and the positioning of the data on the output report.

The NEXT GROUP clause can only be specified in the level 01 entry. It indicates the line spacing that occurs after the report group being described is written on the file and before the next report group is processed. When the next report group is processed, line spacing specified for that report group is in addition to the line spacing of the NEXT GROUP clause. The line spacing can be specified in the NEXT GROUP clause as an absolute line number, a relative line number, or the next page. An absolute line number causes the next report group to be written on the specified line number. A relative line number increments the line counter for the report by the specified number before processing the next report group. If NEXT PAGE is specified, a page eject occurs before the next report group is written on the report file. When the PAGE clause is not specified in the RD entry for the report, only relative line spacing can be specified in the NEXT GROUP clause.

The LINE NUMBER clause is included once for each print line of the report group. If only one line is to be printed, the clause can be specified in the level 01 entry; otherwise, it is



Figure 3-1. Report Worksheet for SALESRP Program

TABLE 3-1. SUMMARY OF PAGE AREA LIMITS

PAGE AREA	FIRST LINE	LAST LINE
Heading area	HEADING line number; default is line number 1.	First line of body area minus 1.
Body area	FIRST DETAIL line number; default is first line of heading area.	FOOTING line number, if specified; otherwise, LAST DETAIL line number; default is PAGE LIMIT line number.†
Footing area	Last line of body area plus 1.	PAGE LIMIT line number.

†When LAST DETAIL and FOOTING are specified, detail lines are not written on lines following the LAST DETAIL line number.

```
RD  SALES
        PAGE LIMIT IS 50 LINES
            HEADING      3
            FIRST DETAIL 10
            LAST DETAIL  40.
```

Figure 3-2. RD Entry for SALES Report

pecified in either group or elementary subordinate entries. An absolute line number indicates that the report group rint line appears on the specified line number; if the NEXT 'AGE phrase follows the absolute line number, a page eject ccurs before the print line is presented on the specified line umber. Absolute line numbers can be specified only when he RD entry includes the PAGE clause. Each absolute line umber specified must be greater than the line number in he report group previously processed. A relative line umber is designated by the key word PLUS followed by a umber. The line counter is incremented by the specified umber before the print line is written on the report file.

he information to be printed on a line is specified in one or ore elementary entries in the Report Group Description ntry. The PICTURE clause must be specified to describe he general characteristics and editing requirements of the em to be printed. The COLUMN NUMBER clause specifies he number of the column that is the leftmost character osition of the item to be printed. When more than one lementary entry defines the items on a print line, the OLUMN NUMBER clauses must specify column numbers in scending order.

he specific data to be printed is identified by either the OURCE clause or the VALUE clause. The SOURCE clause pecifies the data-name of an item that is moved to the rint line in the same manner as if a MOVE statement had een executed. The data item can be an item described in any other section of the Data Division or the special register LINE-COUNTER or PAGE-COUNTER. The VALUE clause specifies a constant value that is printed each time the report group is generated.

REPORT HEADING

A report heading appears only on the first page of the report. Only one report heading report group can be specified for a report. The report group is processed and written on the report file during the first execution of a GENERATE statement for the report.

The level 01 entry contains a TYPE clause that specifies REPORT HEADING or its abbreviation RH. The NEXT GROUP clause, if included in the entry, can affect the positioning of a page heading report group. If NEXT GROUP NEXT PAGE is specified, the report heading is the only report group printed on the first page of the report.

The vertical positioning of a line in the report heading is determined by the LINE NUMBER clause. An absolute line number causes the print line to appear on the specified line number. If the PAGE clause is included in the RD entry, the line number must be within the specified heading area of the page. When the first LINE NUMBER clause in the report heading report group specifies a relative line number, Report Writer calculates the line number based on the presence or absence of the PAGE clause. If the PAGE clause is specified, the relative line number is added to a value that is equal to one less than the first line number of the heading area. If the PAGE clause is omitted, the value of LINE-COUNTER is added to the relative line number; the value of LINE-COUNTER is zero when the report heading is being generated.

The Report Group Description entry shown in figure 3-3 describes the report heading for the SALES report. The report heading consists of two lines. Each line is described as a level 03 elementary item. The first print line contains

```
01   TYPE IS REPORT HEADING.
     03   LINE NUMBER 3   COLUMN NUMBER 57   PICTURE X(24)
               VALUE IS "SALES REPORT BY SALESMAN".
     03   LINE NUMBER 4   COLUMN NUMBER 62   PICTURE X(14)
               SOURCE IS DATE-IN.
```

Figure 3-3. Report Heading Report Group Description Entry for SALES Report

the title of the report beginning in column number 57 of line number 3; the VALUE clause specifies the actual title. The second line of the report heading begins in column number 62 of line number 4; the SOURCE clause specifies that the current value of the data item DATE-IN is moved to the designated position when the report heading report group is processed. These two lines are printed only on the first page of the report.

PAGE HEADING

A page heading appears at the beginning of each page. Unless a report heading or report footing report group is designated to appear on a page by itself, the page heading report group is written on every page of the report. Only one page heading report group can be specified for a report; however, more than one line can be described in the Report Group Description entry. The PAGE clause must be included in the RD entry when a page heading report group is specified.

The TYPE clause in the level-01 entry specifies PAGE HEADING or its abbreviation PH. The NEXT GROUP clause cannot be specified for a page heading report group.

A line of the page heading is positioned on the page according to the LINE NUMBER clause. If an absolute line number is specified, the line appears on the designated line number, which must be within the heading area of the report. When the first LINE NUMBER clause in the page heading report group is a relative line number and a report heading is on the same page, the value of LINE-COUNTER is incremented by the specified number. For a page on which the page heading is the first report group, the relative line number is added to a value that is one less than the first line number of the heading area.

The page heading Report Group Description entry for the SALES report is shown in figure 3-4. Each of the two lines is described as a level 03 group item. Relative line numbers are specified for both lines. On the first page of the report, the report heading appears before the page heading; therefore, the first line of the page heading appears on line number 7 (LINE-COUNTER equals 4 and is incremented by 3). On succeeding pages of the report, the first line of the page heading appears on line number 5 (2, which is the number of the line preceding the heading area, plus 3, which is the relative line number). The COLUMN NUMBER clauses specify the columns in which the nonnumeric literals in the VALUE clauses are to begin.

DETAIL LINE

A detail line is written on the report file during execution of a GENERATE statement that specifies the data-name of the detail report group. More than one detail report group can be described for a report. A GENERATE statement is required for each detail report group that is to be written on the report file.

When a summary report is generated, detail lines are not written on the report file. The GENERATE statement for a summary report specifies the report-name rather than the data-name of a detail report group. If a summary report is being generated, only one detail report group can be described; however, a detail report group is not required.

The level 01 entry for a detail report group contains a TYPE clause that specifies DETAIL or its abbreviation DE. A data-name must be specified if the detail line is to be printed on the report; the GENERATE statement specifies the data-name of the detail report group.

The NEXT GROUP clause, if specified for the detail report group, refers to the positioning of the report group processed immediately following the detail line. When this clause is specified, LINE-COUNTER is incremented to reflect the indicated line spacing after the last line of the detail report group is written on the report file.

Line spacing for the detail report group is specified by the LINE NUMBER clause. If an absolute line number is specified, the detail report group appears only once on a page; an absolute line number cannot be specified when a relative line number is specified for the report group processed immediately before the detail report group. A relative line number is usually specified for a detail report group. This allows the detail report group to be presented a number of times on each page of the report.

Line spacing for the first line of a detail report group with a relative line number depends on whether or not the PAGE clause is included in the RD entry for the report. If the PAGE clause is omitted, the line number is the current value of LINE-COUNTER plus the relative line number. When the PAGE clause is specified, the incrementing of LINE-COUNTER occurs as follows:

If the value of LINE-COUNTER is less than the first line number of the body area, LINE-COUNTER is incremented to indicate the first line of the body area.

If the value of LINE-COUNTER indicates a line within the body area and the report group is being presented for the first time on the page, LINE-COUNTER is incremented by 1.

```
01   TYPE IS PAGE HEADING.
     03   LINE NUMBER PLUS 3.
          05   COLUMN NUMBER 21    PICTURE X(8)   VALUE IS "SALESMAN".
          05   COLUMN NUMBER 41    PICTURE X(5)   VALUE IS "SALES".
          05   COLUMN NUMBER 61    PICTURE X(4)   VALUE IS "THIS".
          05   COLUMN NUMBER 85    PICTURE X(4)   VALUE IS "LAST".
          05   COLUMN NUMBER 110   PICTURE XXX    VALUE IS "YTD".
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 22    PICTURE X(6)   VALUE IS "NUMBER".
          05   COLUMN NUMBER 39    PICTURE X(9)   VALUE IS "TERRITORY".
          05   COLUMN NUMBER 61    PICTURE X(5)   VALUE IS "MONTH".
          05   COLUMN NUMBER 85    PICTURE X(5)   VALUE IS "MONTH".
          05   COLUMN NUMBER 109   PICTURE X(5)   VALUE IS "SALES".
```

Figure 3-4. Page Heading Report Group Description Entry for SALES Report

If the value of LINE-COUNTER indicates a line within the body area and the report group is not being presented for the first time on the page, LINE-COUNTER is incremented by the relative line number.

The GROUP INDICATE clause can be specified for a printable item in a detail line. When this clause is specified and the PAGE clause is included in the RD entry, the item is printed the first time the detail line is written on each page; if the PAGE clause is omitted, the item is printed only the first time the detail report group is processed. This clause can be used for a printable item that remains constant for the entire report and only needs to be printed once per page. Refer to section 4 for the effect of control breaks on the GROUP INDICATE clause.

Figure 3-5 shows the Report Group Description entry for the detail lines generated for the SALES report. The data-name DET-LINE is used in the GENERATE statement for the detail report group. A relative line number, PLUS 2, is specified for the detail line. The first detail line is written on line number 10, which is the first line of the body area. LINE-COUNTER is then incremented by 2 each time a detail line is generated, thus producing a double spaced report. Each of the five level 03 elementary entries specifies a data item to be written on the report beginning in the designated column number. The data items THIS-MONTH, LAST-MONTH, and YTD are edited according to their respective PICTURE clause specification before the detail line is written on the report file.

PAGE FOOTING

A page footing report group appears at the end of each page. If a report heading or a report footing is designated to appear on a page by itself, the page footing is not written on that page. The only report group that can follow the page footing is the report footing, which is written only on the last page of the report. A page footing report group can be specified only if the PAGE clause is included in the RD entry for the report, and then only one page footing report group can be described.

The level 01 entry contains a TYPE clause that specifies PAGE FOOTING or its abbreviation PF. The NEXT GROUP clause can be included to specify positioning of the report footing report group. Either an absolute line number or a relative line number can be specified. The line number in the NEXT GOUP clause must be within the footing area defined for the report page.

A LINE NUMBER clause is specified for each line described in the page footing report group. The first line number must be an absolute line number; either absolute or relative line numbers can be specified for additional lines in the report group. Any line number in the page footing report group must be within the footing area.

The Report Group Description entry shown in figure 3-6 describes the page footing for the SALES report. The page footing appears on line number 44 on each page of the report. The special register PAGE-COUNTER is specified as the source item for the page number to be written on each page.

REPORT FOOTING

A report footing is the last report group written on the report file during report generation. It appears only on the last page of the report. The report footing report group is processed and written on the report file during execution of the TERMINATE statement. Only one report footing report group can be specified for a report; however, more than one line can be described in the report group.

The level 01 entry contains a TYPE clause that specifies REPORT FOOTING or its abbreviation RF. The NEXT GROUP clause is not included because the report footing is the last report group presented on the report.

The positioning of the report footing on the report page is determined by the LINE NUMBER clause for the first line of the report group. When an absolute line number is specified and the NEXT PAGE phrase is included, the report footing can appear anywhere on the page from the first line of the heading area to the specified page limit; otherwise, the report footing must be written within the footing area defined for the page. A relative line number causes LINE-COUNTER to be incremented by the specified number. If the value of LINE-COUNTER indicates a line number greater than the specified page limit, a page advance occurs. The report footing is then written on the line number that is equal to the last line number of the body area plus the relative line number.

```
01   DET-LINE   TYPE IS DETAIL   LINE NUMBER PLUS 2.
       03   COLUMN NUMBER 23    PICTURE X(4)   SOURCE IS SALESMAN.
       03   COLUMN NUMBER 42    PICTURE XX     SOURCE IS TERRITORY.
       03   COLUMN NUMBER 58    PICTURE $$$,999.99
                 SOURCE IS THIS-MONTH.
       03   COLUMN NUMBER 82    PICTURE $$$,999.99
                 SOURCE IS LAST-MONTH.
       03   COLUMN NUMBER 106   PICTURE $$$$,999.99
                 SOURCE IS YTD.
```

Figure 3-5. Detail Report Group Description Entry for SALES Report

```
01   TYPE IS PAGE FOOTING        LINE NUMBER 44.
       03   COLUMN NUMBER 109   PICTURE X(4)   VALUE IS "PAGE".
       03   COLUMN NUMBER 114   PICTURE ZZ9    SOURCE IS PAGE-COUNTER.
```

Figure 3-6. Page Footing Report Group Description Entry for SALES Report

The report footing for the SALES report is described by the Report Group Description entry shown in figure 3-7. A relative line number is specified for the report footing. When the report is terminated, LINE-COUNTER is incremented by 4 before the report footing is written on the report file. The report footing follows the page footing on the same page.

SALESRP PROGRAM

The report that has been described in this section is generated by the SALESRP program shown in figure 3-8. This program reads input records and generates detail lines from the input data.

The report file (OUT-FILE) is assigned to the system file OUTPUT (line 10); therefore, the report is printed at program termination. The File Description entry for OUT-FILE specifies the report-name SALES (line 29). The Report Section consists of the RD entry and Report Group Description entries previously discussed in this section.

The data item DATE-IN is accepted from the system file INPUT (line 76). This data item is the source item for the report heading (line 42) and for the report footing (line 71). The INITIATE statement (line 77) sets the special registers LINE-COUNTER and PAGE-COUNTER to 0 and 1, respectively. For each input card read (line 79), a detail line is generated (line 81) according to the report specifications. Headings and footings are automatically generated at the appropriate times. When all input cards have been read, the TERMINATE statement is executed (line 84). This statement causes the page footing for the last page and the report footing to be written on the report file.

Figure 3-9 shows the input data used to create the SALES report shown in figure 3-10.

```
01   TYPE IS REPORT FOOTING   LINE NUMBER PLUS 4.
03   COLUMN NUMBER 40   PICTURE X(43)   VALUE IS
         "THIS COMPLETES THE MONTHLY SALES REPORT FOR".
03   COLUMN NUMBER 84   PICTURE X(14)   SOURCE IS DATE-IN.
```

Figure 3-7. Report Footing Report Group Description Entry for SALES Report

```
 1   IDENTIFICATION DIVISION.
 2   PROGRAM-ID.  SALESRP.
 3   ENVIRONMENT DIVISION.
 4   CONFIGURATION SECTION.
 5   SOURCE-COMPUTER.  CYBER-170.
 6   OBJECT-COMPUTER.  CYBER-170.
 7   INPUT-OUTPUT SECTION.
 8   FILE-CONTROL.
 9       SELECT IN-FILE  ASSIGN TO ≠INPUT≠.
10       SELECT OUT-FILE ASSIGN TO ≠OUTPUT≠.
11   DATA DIVISION.
12   FILE SECTION.
13   FD  IN-FILE
14           LABEL RECORDS ARE OMITTED
15           DATA RECORD IS IN-REC.
16   01  IN-REC.
17       03  SALESMAN              PICTURE 9(4).
18       03  FILLER                PICTURE XXX.
19       03  TERRITORY             PICTURE XX.
20       03  FILLER                PICTURE XXX.
21       03  THIS-MONTH            PICTURE 9(5)V99.
22       03  FILLER                PICTURE XXX.
23       03  LAST-MONTH            PICTURE 9(5)V99.
24       03  FILLER                PICTURE XXX.
25       03  YTD                   PICTURE 9(6)V99.
26       03  FILLER                PICTURE X(40).
27   FD  OUT-FILE
28           LABEL RECORDS ARE OMITTED
29           REPORT IS SALES.
30   WORKING-STORAGE SECTION.
31   77  DATE-IN                   PICTURE X(14).
32   REPORT SECTION.
33   RD  SALES
34           PAGE LIMIT IS 50 LINES
35               HEADING      3
36               FIRST DETAIL 10
37               LAST DETAIL  40.
38   01  TYPE IS REPORT HEADING.
39       03  LINE NUMBER 3  COLUMN NUMBER 57   PICTURE X(24)
```

Figure 3-8. SALESRP Program (Sheet 1 of 2)

```
40                  VALUE IS ≠SALES REPORT BY SALESMAN≠.
41          03  LINE NUMBER 4  COLUMN NUMBER 62   PICTURE X(14)
42                  SOURCE IS DATE-IN.
43      01  TYPE IS PAGE HEADING.
44          03  LINE NUMBER PLUS 3.
45              05  COLUMN NUMBER 21  PICTURE X(8)  VALUE IS ≠SALESMAN≠.
46              05  COLUMN NUMBER 41  PICTURE X(5)  VALUE IS ≠SALES≠.
47              05  COLUMN NUMBER 61  PICTURE X(4)  VALUE IS ≠THIS≠.
48              05  COLUMN NUMBER 85  PICTURE X(4)  VALUE IS ≠LAST≠.
49              05  COLUMN NUMBER 110 PICTURE XXX   VALUE IS ≠YTD≠.
50          03  LINE NUMBER PLUS 1.
51              05  COLUMN NUMBER 22  PICTURE X(6)  VALUE IS ≠NUMBER≠.
52              05  COLUMN NUMBER 39  PICTURE X(9)  VALUE IS ≠TERRITORY≠.
53              05  COLUMN NUMBER 61  PICTURE X(5)  VALUE IS ≠MONTH≠.
54              05  COLUMN NUMBER 85  PICTURE X(5)  VALUE IS ≠MONTH≠.
55              05  COLUMN NUMBER 109 PICTURE X(5)  VALUE IS ≠SALES≠.
56      01  DET-LINE  TYPE IS DETAIL  LINE NUMBER PLUS 2.
57          03  COLUMN NUMBER 23  PICTURE X(4)  SOURCE IS SALESMAN.
58          03  COLUMN NUMBER 42  PICTURE XX    SOURCE IS TERRITORY.
59          03  COLUMN NUMBER 58  PICTURE $$$,999.99
60                  SOURCE IS THIS-MONTH.
61          03  COLUMN NUMBER 82  PICTURE $$$,999.99
62                  SOURCE IS LAST-MONTH.
63          03  COLUMN NUMBER 106 PICTURE $$$$,999.99
64                  SOURCE IS YTD.
65      01  TYPE IS PAGE FOOTING  LINE NUMBER 44.
66          03  COLUMN NUMBER 109 PICTURE X(4)  VALUE IS ≠PAGE≠.
67          03  COLUMN NUMBER 114 PICTURE ZZ9   SOURCE IS PAGE-COUNTER.
68      01  TYPE IS REPORT FOOTING  LINE NUMBER PLUS 4.
69          03  COLUMN NUMBER 40  PICTURE X(43)  VALUE IS
70                  ≠THIS COMPLETES THE MONTHLY SALES REPORT FOR≠.
71          03  COLUMN NUMBER 84  PICTURE X(14)  SOURCE IS DATE-IN.
72      PROCEDURE DIVISION.
73      OPENING.
74          OPEN INPUT IN-FILE.
75          OPEN OUTPUT OUT-FILE.
76          ACCEPT DATE-IN.
77          INITIATE SALES.
78      READING.
79          READ IN-FILE RECORD
80              AT END GO TO CLOSING.
81          GENERATE DET-LINE.
82          GO TO READING.
83      CLOSING.
84          TERMINATE SALES.
85          CLOSE IN-FILE, OUT-FILE.
86          STOP RUN.
```

Figure 3-8. SALESRP Program (Sheet 2 of 2)

```
                JUNE 1976
Column 1        Column 8    Column 13   Column 23   Column 33
1062    SW      0946538     1145793     04966515
1436    NE      1477209     1356987     06531048
1899    SO      0832145     0622514     03876225
2275    SE      1129516     0944820     05248996
2361    SW      1045488     1062354     05112459
3124    NO      0984506     0925061     04834103
3865    SE      0893342     1129845     05061764
4013    NW      1216590     1245622     06144378
4280    NE      1604377     1537365     07292015
4949    SO      0956879     0998351     04805633
5506    NO      1301695     1087626     06318510
5772    NE      1525483     1290418     06824038
6311    NW      0791240     0853862     04465127
6578    NO      1076823     1149075     05892268
7092    SW      0935452     0998760     04736443
7244    NE      1287860     0991034     05579125
7505    SE      1402930     1367803     06790434
7890    SW      0889645     1039366     04588720
8321    NO      1310523     1287000     06913254
8626    SO      1042916     0956241     04806126
9178    SE      1685243     1480683     07924355
9787    NW      0991078     1145720     05793810
```

Figure 3-9. Input Data for SALESRP Program

SALES REPORT BY SALESMAN
JUNE 1976

SALESMAN NUMBER	SALES TERRITORY	THIS MONTH	LAST MONTH	YTD SALES
1062	SW	$9,465.38	$11,457.93	$49,665.15
1436	NE	$14,772.09	$13,569.87	$65,310.48
1899	SO	$8,321.45	$6,225.14	$38,762.25
2275	SE	$11,295.16	$9,448.20	$52,489.96
2361	SW	$10,454.88	$10,623.54	$51,124.59
3124	NO	$9,845.06	$9,250.61	$48,341.03
3865	SE	$8,933.42	$11,298.45	$50,617.64
4013	NW	$12,165.90	$12,456.22	$61,443.78
4280	NE	$16,043.77	$15,373.65	$72,920.15
4949	SO	$9,568.79	$9,983.51	$48,056.33
5506	NO	$13,016.95	$10,876.26	$63,185.10
5772	NE	$15,254.83	$12,904.18	$68,240.38
6311	NW	$7,912.40	$8,538.62	$44,651.27
6578	NO	$10,768.23	$11,490.75	$58,922.68
7092	SW	$9,354.52	$9,987.60	$47,364.43
7244	NE	$12,878.60	$9,910.34	$55,791.25

PAGE 1

Figure 3-10. SALES Report Generated by SALESRP Program (Sheet 1 of 2)

SALESMAN NUMBER	SALES TERRITORY	THIS MONTH	LAST MONTH	YTD SALES
7505	SE	$14,029.30	$13,678.03	$67,904.34
7890	SW	$8,896.45	$10,393.66	$45,887.20
8321	NO	$13,105.23	$12,870.00	$69,132.54
8626	SO	$10,429.16	$9,562.41	$48,061.26
9178	SE	$16,852.43	$14,806.83	$79,243.55
9787	NW	$9,910.78	$11,457.20	$57,938.10

PAGE 2

THIS COMPLETES THE MONTHLY SALES REPORT FOR JUNE 1976

Figure 3-10. SALES Report Generated by SALESRP Program (Sheet 2 of 2)

A technique often used in generating reports is to interrupt the printing of detail lines in the body of the report to print footing and heading lines. The interruption occurs when the value of a control data item changes. The control data item that is monitored for a change in value is called a break control item; the interruption is called a control break. Control headings and control footings are generated automatically by Report Writer whenever a control break occurs during the processing of a detail line. Control footings can include totals accumulated as detail lines are processed; these totals are automatically accumulated by Report Writer.

Two programs are presented in this section to illustrate the use of control breaks. The first program, TERRTRY, specifies a control heading and a control footing that appear repeatedly within the body of the report as well as a final control footing that is the last body report group presented. The second program, BUDGETS, illustrates the use of four levels of control breaks in the report.

IDENTIFYING THE BREAK CONTROL ITEMS

One or more data items can be monitored for a control break. In addition, a FINAL control break can be specified; this allows a control heading and a control footing to be printed as the first and the last report groups within the body of the report. The CONTROL clause in the RD entry for the report specifies the break control items. Control breaks are not processed unless this clause is specified.

A FINAL control break is specified by including the key word FINAL in the CONTROL clause. If a FINAL control break is to be processed, it must be specified first in the clause. A control heading and/or a control footing report group can be described for the FINAL control break. A control heading for the FINAL control break is the first report group within the body area on the first page of the report. This type of report group is specified for a heading that appears only on the first page and immediately follows the page heading. A control footing for the FINAL control break is the last report group within the body area of the last page of the report. This type of report group can be used when a final total is to be written on the last page.

Data items that are defined in the File Section, Working-Storage Section, or Common-Storage Section can be specified as break control items. The data items must be fixed-length items. A data-name specified in the CONTROL clause can be qualified; however, it cannot be subscripted or indexed.

When the first GENERATE statement is executed, the value of each break control item is saved. These values, which are called prior values, are used to check for a control break during subsequent executions of the GENERATE statement. When a control break occurs, the prior values are made available for source data items in control footing report groups and for processing in a declarative procedure; declarative procedures are discussed in section 6. After the control break is processed, the new break control item values are saved as the prior values.

In the TERRTRY program, the CONTROL clause for the TERRITORIES report specifies two break control items: FINAL and TERRITORY. This clause is shown in figure 4-1. At least one control report group (footing or heading) is described for each of the break control items. The TERRITORIES report describes a final control footing; therefore, the FINAL control break is processed as the last control break of the report and the final control footing is the last body group written on the report file. The TERRITORY control break is processed each time the value of the data item TERRITORY changes; TERRITORY is a data item defined in the File Section of the Data Division.

```
RD   TERRITORIES
          CONTROLS ARE FINAL, TERRITORY
          PAGE LIMIT IS 55 LINES
               HEADING       3
               FIRST DETAIL  7
               LAST DETAIL  47
               FOOTING      50.
```

Figure 4-1. RD Entry for TERRITORIES Report

ESTABLISHING THE HIERARCHY OF CONTROL BREAKS

The order in which control headings and control footings are presented in the report is determined by the order in which the break control items are specified in the CONTROL clause. The break control items are specified from high level to low level. The first data item specified is the major break control item; the last data item specified is the minor break control item. Control headings are written on the report file from high level to low level; control footings are written on the report file from low level to high level.

When the first GENERATE statement is executed, the control headings associated with the break control items are written on the report file beginning with the control heading for the major break control item and ending with the control heading for the minor break control item. Detail lines are then generated until a control break occurs. At this time, the control footings are written on the report file beginning with the control footing associated with the minor break control item and ending with the control footing associated with the break control item that caused the control break. Control headings are then written on the report beginning with the heading associated with the break control item that caused the control break and ending with the heading associated with the minor break control item.

The CONTROL clause in the BUDGETS program is shown in figure 4-2, the RD entry for the EXPENSES report. Four break control items are specified: FINAL, DEPT, SECT, and ACCOUNT. A FINAL control heading and a FINAL control footing are described for this report; the FINAL control break is processed for the first and the last body report groups. Each time a GENERATE statement is executed, the DEPT, SECT, and ACCOUNT data items are checked for a

```
RD EXPENSES
        CONTROLS ARE FINAL, DEPT, SECT,
            ACCOUNT
        PAGE LIMIT IS 55  LINES
            HEADING          2
            FIRST DETAIL     7
            LAST DETAIL     50
            FOOTING         52.
```

Figure 4-2. RD Entry for EXPENSES Report

change in value. When a control break occurs at the DEPT level, control footings are produced for ACCOUNT, SECT, and then DEPT. Control headings are then written for DEPT, SECT, and ACCOUNT. After the ACCOUNT control heading is written, the detail line is written on the report file.

SPECIFYING THE PAGE LIMITS

Control report groups are presented within the body area of a report page. The body area is defined in the PAGE clause by the FIRST DETAIL, LAST DETAIL, and FOOTING phrases.

When all three of the phrases are specified in the PAGE clause, control heading and detail report groups are written only on the lines beginning with the FIRST DETAIL line number and ending with the LAST DETAIL line number. Control footing report groups can be written anywhere within the area specified by the FIRST DETAIL and FOOTING phrases. The last line of a control footing report group can appear on the line specified by the FOOTING phrase.

If only the FIRST DETAIL and LAST DETAIL phrases are specified, the line number of the LAST DETAIL phrase is the last line on which a control footing report group can appear. Specifying FIRST DETAIL and FOOTING allows control heading, detail, and control footing report groups to appear within the specified line numbers. When only the FIRST DETAIL phrase is specified, the body area extends to the page limit.

The PAGE clause for the TERRITORIES report is shown in figure 4-1. Line numbers 7 through 47 define the body area for control headings, detail lines, and control footings. When a control footing is the last report group presented on the page, the control footing can extend to line number 50. The footing area for page and report footings is line numbers 51 through 55.

The RD entry for the EXPENSES report defines a page limit of 55 lines. The heading area consists of lines 2 through 6, the body area consists of lines 7 through 52, and the footing area consists of lines 53 through 55. Within the body area of the report, only control footing report groups can appear on lines 51 and 52.

DESCRIBING THE CONTROL REPORT GROUPS

At least one control report group must be included for each break control item specified in the CONTROL clause for the report. Both a control heading and a control footing can be described for a break control item.

The Report Group Description entry for a control heading or a control footing consists of the same basic elements as the report groups discussed in section 3. The level 01 entry must contain the TYPE clause; it can also include the NEXT GROUP clause. The LINE NUMBER clause is included once for each print line. The PICTURE clause and the COLUMN NUMBER clause are specified for each item to be printed. The SOURCE clause or the VALUE clause can be used to designate the data to be printed; in addition, the SUM clause can be specified for a control footing report group when automatic totaling is to be performed.

The report worksheets for the TERRITORIES report and the EXPENSES report are shown in figures 4-3 and 4-4, respectively.

CONTROL HEADINGS

The report description can include one control heading report group for a FINAL control break and one for each data item specified as a break control item. No control heading report group need be specified for the report. Control headings are presented on the report page according to the hierarchy of control breaks as established by the CONTROL clause.

The TYPE clause in the level 01 entry specifies CONTROL HEADING or its abbreviation CH and either the key word FINAL or a data-name corresponding to a data-name in the CONTROL clause. When FINAL is specified, the report group is processed only during execution of the first GENERATE statement for the report. A control heading for a FINAL control break is the first body report group written on the first page of the report. A control heading for any other control break is written on the report file only when a control break occurs at its level or at a higher level.

If the NEXT GROUP clause is specified for a control heading, the positioning of the report group processed immediately after the control heading is affected. An absolute line number can be specified only if the PAGE clause is included in the RD entry and the report group processed immediately before the control heading specifies an absolute line number.

The LINE NUMBER clause designates the vertical positioning of a line in the control heading. When a relative line number is specified for the first line of the control heading, the line number on which the print line appears is determined by Report Writer based on whether or not the PAGE clause is included in the RD entry for the report. If the PAGE clause is omitted, the current value of LINE-COUNTER is incremented by the relative line number. When the PAGE clause is specified, LINE-COUNTER is incremented as follows:

If LINE-COUNTER indicates a line number in the heading area, LINE-COUNTER is incremented to the first line number of the body area.

If LINE-COUNTER indicates a line number in the body area and the report group is the first body report group on the page, LINE-COUNTER is incremented by 1.

If LINE-COUNTER indicates a line number in the body area and the control heading is not the first body report group on the page, LINE-COUNTER is incremented by the relative line number.

One control heading report group, which is shown in figure 4-5, is specified for the TERRITORIES report. This report group is written on the report file whenever a control break occurs for the data item TERRITORY. The control



Figure 4-3. Report Worksheet for TERRTRY Program



Figure 4-4. Report Worksheet for BUDGETS Program

```
01   TYPE IS CONTROL HEADING TERRITORY   NEXT GROUP PLUS 1.
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 23    PICTURE X(5)   VALUE IS "SALES".
          05   COLUMN NUMBER 40    PICTURE X(8)   VALUE IS "SALESMAN".
          05   COLUMN NUMBER 61    PICTURE X(4)   VALUE IS "LAST".
          05   COLUMN NUMBER 85    PICTURE X(4)   VALUE IS "THIS".
          05   COLUMN NUMBER 110   PICTURE XXX   VALUE IS "YTD".
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 21    PICTURE X(9)   VALUE IS "TERRITORY".
          05   COLUMN NUMBER 41    PICTURE X(6)   VALUE IS "NUMBER".
          05   COLUMN NUMBER 61    PICTURE X(5)   VALUE IS "MONTH".
          05   COLUMN NUMBER 85    PICTURE X(5)   VALUE IS "MONTH".
          05   COLUMN NUMBER 109   PICTURE X(5)   VALUE IS "SALES".
```

Figure 4-5. Control Heading Report Group Description Entry for TERRITORIES Report

heading consists of two lines that print column headings each time the value of TERRITORY changes. When the control heading report group has been processed, LINE-COUNTER is incremented by 1 before processing the next report group.

Figure 4-6 shows the four control heading report groups for the EXPENSES report. A control heading is specified for each of the four break control items designated in the CONTROL clause for the report. The first report group is processed only when the first GENERATE statement is executed for the report. The other three report groups are associated with the break control data items DEPT, SECT, and ACCOUNT. When a control break occurs at the DEPT level, all three report groups are processed. A control break in the SECT data item causes the control headings for SECT and ACCOUNT to be processed. Only the control heading for ACCOUNT is processed when a control break occurs in the ACCOUNT data item.

DETAIL LINES

Detail lines for a report with control breaks are described in the same manner as discussed in section 3. Only the GROUP INDICATE clause is affected by control breaks. When this clause is specified for a printable item, the item is printed the first time the detail report group is processed after each control break as well as after each page advance. Both sample programs shown at the end of this section use the GROUP INDICATE clause to suppress the repeated printing of the minor break control item in detail lines.

CONTROL FOOTINGS

Control footing report groups are included in the report description when footing lines are to be written at the end of a group of detail lines. One control footing report group can be specified for each of the break control items designated in the CONTROL clause for the report. The order in which control footing report groups appear on the report is determined by the hierarchy of control breaks.

The level 01 entry contains a TYPE clause that specifies CONTROL FOOTING or its abbreviation CF and either the key word FINAL or the data-name of a break control item. A control footing report group that specifies FINAL is processed only when the TERMINATE statement is executed. This report group is the last body report group written on the report file. A control footing report group that specifies a data-name is processed whenever a control break occurs at its level or at a higher level.

The NEXT GROUP clause can be included in the level 01 entry of the control footing report group. This clause is processed only when the control break occurs at the level of the break control item associated with the control footing report group. The clause is ignored during processing of the report group when the control break occurs at a higher level. The NEXT GROUP clause can specify an absolute line number, a relative line number, or NEXT PAGE. The positioning of the next report group processed is affected by the NEXT GROUP clause.

```
01   TYPE IS CONTROL HEADING FINAL   LINE NUMBER PLUS 2.
     03   COLUMN NUMBER 19   PICTURE X(16)
               VALUE IS "***REPORT COVERS".
     03   COLUMN NUMBER 36   PICTURE Z9      SOURCE IS MONTH-IN.
     03   COLUMN NUMBER 39   PICTURE X(9)    VALUE IS "MONTHS***".

01   TYPE IS CONTROL HEADING DEPT   LINE NUMBER PLUS 3.
     03   COLUMN NUMBER 22   PICTURE X(10)   VALUE IS "DEPARTMENT".
     03   COLUMN NUMBER 33   PICTURE X(5)    SOURCE IS DEPT.

01   TYPE IS CONTROL HEADING SECT   LINE NUMBER PLUS 2.
     03   COLUMN NUMBER 25   PICTURE X(7)    VALUE IS "SECTION".
     03   COLUMN NUMBER 33   PICTURE X(4)    SOURCE IS SECT.

01   TYPE IS CONTROL HEADING ACCOUNT   LINE NUMBER PLUS 2.
     03   COLUMN NUMBER 33   PICTURE X(7)    VALUE IS "ACCOUNT".
```

Figure 4-6. Control Heading Report Group Description Entries for EXPENSES Report

The vertical positioning of the control footing report group is determined by the LINE NUMBER clause. This clause is specified once for each line in the report group. When a relative line number is specified for the first line of the control footing, Report Writer calculates the line number in the same way as discussed for a control heading report group.

A control footing frequently contains a printable item that is an accumulated total. The total is automatically accumulated by Report Writer when the SUM clause is specified for an item to be printed. When the report group is processed, the accumulated total is written on the control footing print line. The type of totaling that is performed depends on the items being summed.

A sum counter is established for the entry containing the SUM clause. If a data-name is specified for the entry, the sum counter can be referenced in another SUM clause and in Procedure Division statements. The sum counter is a signed numeric data item that is equal in size to the size of the item specified by the PICTURE clause for the entry. When the report group is processed, the value of the sum counter is moved to the printable item as if a MOVE statement had been executed. Any editing specified by the PICTURE clause is performed at this time. The sum counter is set to zero when the INITIATE statement is executed. It is reset to zero after the control footing report group is processed unless the RESET phrase is specified in the SUM clause.

Subtotaling is performed when the SUM clause specifies one or more data items to be added to the sum counter. Whenever a GENERATE statement or the TERMINATE statement is executed, the value of each specified data item is added to the sum counter. The accumulated total is printed when the control footing report group is processed. Subtotaling is used in both the TERRITORIES report and the EXPENSES report.

Crossfooting occurs when the SUM clause specifies one or more sum counters that are defined within the same control footing report group. When the report group is processed as a result of a control break or when the TERMINATE statement is executed, the specified sum counters are added to the sum counter for the entry. Crossfooting can be specified for more than one entry in the report group. When this occurs, crossfooting is performed for the sum counters in the order they are defined in the report group. Crossfooting is used in the EXPENSES report.

Rolling forward occurs when the SUM clause specifies one or more sum counters that are defined in a control footing report group for a lower level control break. Whenever the lower level control break occurs and its associated control footing report group is processed, the value of the specified sum counter is added to the sum counter of the higher level report group. Both the TERRITORIES report and the EXPENSES report roll forward the values of sum counters.

The UPON phrase is specified in the SUM clause when selective subtotaling is to be performed; this phrase cannot be used for crossfooting or rolling forward. Selective subtotaling is used when two or more detail report groups are defined and subtotaling is to be performed only when the specified report groups are processed. The UPON phrase specifies data-names of detail report groups; subtotaling then occurs only when a GENERATE statement for a specified detail report group is executed. If the data-name of a detail report group is specified more than once in the UPON phrase, subtotaling is performed as many times as the data-name is specified. For a report that has three detail lines defined (LINE-1, LINE-2, and LINE-3), the following SUM clause could be specified:

```
SUM ITEM-1 UPON LINE-1, LINE-3
```

Whenever the GENERATE LINE-1 or GENERATE LINE-3 statement is executed, the value of ITEM-1 is added to the sum counter. Execution of the GENERATE LINE-2 statement, however, does not affect the sum counter.

The RESET phrase is specified in the SUM clause to inhibit resetting the sum counter to zero until a designated control break occurs. The specified control break must be a higher level control break than the one associated with the control footing report group containing the RESET phrase. When the RESET phrase is specified, the sum counter produces a cumulative total each time the report group is processed until such time that the designated control break occurs. The sum counter is then reset to zero and the accumulation begins again. If the RESET phrase is not specified, the sum counter is reset to zero each time the report group is processed.

The two control footing report groups for the TERRITORIES report are shown in figure 4-7. The first report group is processed each time a control break occurs in the TERRITORY data item. Three sum counters are specified for the second line of the control footing. Subtotaling is performed for each of the data items LAST-MONTH, THIS-MONTH, and YTD whenever a GENERATE statement is executed. When a TERRITORY control break is detected, the report group is written on the report file. The TERRIROTY prior value, which is the value used to detect the control break, is the value that is used as the source item for the report group. The sum counters are reset to zero, and LINE-COUNTER is incremented by 6 before the next report group is processed.

The second control footing report group shown in figure 4-7 is written only once and is the last body report group written on the report file. The second line of the FINAL control footing defines three sum counters. Each SUM clause specifies a sum counter in the TERRITORY control footing report group. Whenever the TERRITORY report group is processed, the values of the sum counters TOT-1, TOT-2, and TOT-3 are rolled forward; each sum counter is added to a sum counter defined in the FINAL control footing report group.

Figure 4-8 shows the four control footing report groups for the EXPENSES report. One control footing is associated with each of the four control breaks FINAL, DEPT, SECT, and ACCOUNT. A total of eight sum counters are defined in the report groups. Subtotaling, crossfooting, and rolling forward are performed for the control footings.

The ACCOUNT control footing report group describes three lines that are written on the report when a control break occurs in the ACCOUNT data item or in a higher level control break data item. The two sum counters in the second line, BUD-1 and ACT-1, produce a subtotal for the two data items BUDGET and ACTUAL. After the report group is processed, these sum counters are reset to zero. The last line for the ACCOUNT control break also contains two sum counters, BUD-2 and ACT-2. When the report group is processed, crossfooting occurs for these sum counters; the BUD-1 and ACT-1 sum counters are added to the BUD-2 and ACT-2 sum counters, respectively. The BUD-2 and ACT-2 sum counters are not reset to zero until the report group is processed as a result of a control break in the SECT data item; therefore, BUD-2 and ACT-2 contain cumulative totals for a section each time the ACCOUNT control footing report group is processed.

```
01   TYPE IS CONTROL FOOTING TERRITORY   NEXT GROUP PLUS 6.
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 57    PICTURE X(60)   VALUE IS ALL "-".
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 21    PICTURE X(26)
                    VALUE IS "TOTALS FOR SALES TERRITORY".
          05   COLUMN NUMBER 48    PICTURE XX  SOURCE IS TERRITORY.
          05   TOT-1  COLUMN NUMBER 57    PICTURE $$$$,999.99
                    SUM LAST-MONTH.
          05   TOT-2  COLUMN NUMBER 81    PICTURE $$$$,999.99
                    SUM THIS-MONTH.
          05   TOT-3  COLUMN NUMBER 104   PICTURE $$,$$$,999.99
                    SUM YTD.

01   TYPE IS CONTROL FOOTING FINAL.
     03   LINE NUMBER PLUS 2.
          05   COLUMN NUMBER 55    PICTURE X(62)   VALUE IS ALL "-".
     03   LINE NUMBER PLUS 2.
          05   COLUMN NUMBER 21    PICTURE X(32)
                    VALUE IS "TOTALS FOR ALL SALES TERRITORIES".
          05   COLUMN NUMBER 55    PICTURE $$,$$$,999.99   SUM TOT-1.
          05   COLUMN NUMBER 79    PICTURE $$,$$$,999.99   SUM TOT-2.
          05   COLUMN NUMBER 103   PICTURE $$$,$$$,999.99  SUM TOT-3.
```

Figure 4-7. Control Footing Report Group Description Entries for TERRITORIES Report

```
01   TYPE IS CONTROL FOOTING ACCOUNT.
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 49    PICTURE X(10)   VALUE IS ALL "-".
          05   COLUMN NUMBER 89    PICTURE X(10)   VALUE IS ALL "-".
     03   LINE NUMBER PLUS 1.
          05   BUD-1  COLUMN NUMBER 49   PICTURE $$$,999.99
                    SUM BUDGET.
          05   ACT-1  COLUMN NUMBER 89   PICTURE $$$,999.99
                    SUM ACTUAL.
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 25   PICTURE X(20)
                    VALUE IS "SECTION ACCUM TOTALS".
          05   BUD-2  COLUMN NUMBER 68   PICTURE $$$$,999.99
                    SUM BUD-1 RESET ON SECT.
          05   ACT-2  COLUMN NUMBER 108 PICTURE $$$$,999.99
                    SUM ACT-1 RESET ON SECT.

01   TYPE IS CONTROL FOOTING SECT   LINE NUMBER PLUS 2.
     03   COLUMN NUMBER 22    PICTURE X(23)
               VALUE IS "DEPARTMENT ACCUM TOTALS".
     03   COLUMN NUMBER 66    PICTURE $$,$$$,999.99
               SUM BUD-1 RESET ON DEPT.
     03   COLUMN NUMBER 106   PICTURE $$,$$$,999.99
               SUM ACT-1 RESET ON DEPT.

01   TYPE IS CONTROL FOOTING DEPT   LINE NUMBER PLUS 2
          NEXT GROUP NEXT PAGE.
     03   COLUMN NUMBER 19   PICTURE X(100)   VALUE ALL "*".

01   TYPE IS CONTROL FOOTING FINAL.
     03   LINE NUMBER PLUS 4.
          05   COLUMN NUMBER 16    PICTURE X(106)   VALUE IS ALL "*".
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 19    PICTURE X(30)
                    VALUE IS "GRAND TOTALS - ALL DEPARTMENTS".
          05   COLUMN NUMBER 65    PICTURE $$$,$$$,999.99
                    SUM BUD-1 RESET ON FINAL.
          05   COLUMN NUMBER 105   PICTURE $$$,$$$,999.99
                    SUM ACT-1 RESET ON FINAL.
     03   LINE NUMBER PLUS 1.
          05   COLUMN NUMBER 16    PICTURE X(106)   VALUE IS ALL "*".
```

Figure 4-8. Control Footing Report Group Description Entries for EXPENSES Report

The SECT control footing report group is processed when a control break occurs at the SECT level. The two sum counters in this report group produce cumulative BUDGET and ACTUAL totals for sections within a department. Each time the ACCOUNT control footing report group is processed, the BUD-1 and ACT-1 sum counters are rolled forward and added to the sum counters defined in this report group. These sum counters are not reset to zero until the control break occurs at the DEPT level.

The DEPT control footing report group causes a line of asterisks to be written on the report file. This report group is processed after all sections within a department have been listed on the report; that is, it is processed when a control break occurs in the DEPT data item. The NEXT GROUP clause specifies that the next report group processed is written on the next page of the report; therefore, the listing for each department begins on a new page.

The FINAL control footing report group is processed only once. It is the last body report group written on the report. Two sum counters are defined in this report group. When the FINAL control footing report group is processed, the sum counters contain cumulative totals of the BUDGET and ACTUAL data items for all departments listed on the report. The BUD-1 and ACT-1 sum counters for the ACCOUNT control footing report group are rolled forward and added to the sum counters in this report group whenever the ACCOUNT control footing report group is processed.

TERRTRY PROGRAM

The TERRTRY program, which is shown in figure 4-9, generates the TERRITORIES report. This program uses the same input data as the sample program in section 3; however, the input records are in order by sales territory rather than by salesman number. As each input record is read, the TERRITORY data item is checked for a control break. When a control break is detected, totals are printed for the sales territory.

The input and output files are assigned to the system files INPUT and OUTPUT (lines 9 and 10). The File Description entry for the output file specifies the report-name TERRITORIES (line 29). The RD entry and the control Report Group Description entries for the report have already been discussed in this section. The Report Section also describes a page heading report group (lines 40 through 44), a detail report group (lines 58 through 67), and a page footing report group (lines 90 through 94). The detail report group includes the GROUP INDICATE clause for the printable item TERRITORY (lines 59 and 60). The TERRITORY data item is printed the first time the detail report group is processed after a control break and after a page advance.

When the INITIATE statement is executed (line 100), the special registers LINE-COUNTER and PAGE-COUNTER are set to 0 and 1, respectively. The first time the GENERATE statement is executed (line 104), the page heading and control heading report groups are written on the report file before the detail line is generated. For each subsequent execution of the GENERATE statement, the current value of the TERRITORY data item is compared with the previous value of TERRITORY. When a change in value occurs, the TERRITORY control footing and the TERRITORY control heading report groups are processed; the detail line is then written on the report file.

The TERMINATE statement (line 107) is executed when the end of the input file is reached. This statement causes the TERRITORY control footing report group to be processed as if a control break had occurred. The FINAL control footing report group is then written as the last body report group of the report and the page footing is written for the last page.

The input data illustrated in figure 4-10 generates the TERRITORIES report shown in figure 4-11.

BUDGETS PROGRAM

The EXPENSES report that has been discussed in this section is generated by the BUDGETS program shown in figure 4-12. Four control breaks are specified for the EXPENSES report. The FINAL control break is processed at the beginning of the report and at the end of the report. The other three break control items cause the printing of detail lines to be interrupted in order to print accumulated totals for each account, for all accounts within a section, and for all sections within a department.

The input file is assigned to the system file INPUT (line 9) and contains the data items used to generate the report (lines 16 through 26). The output file is assigned to the system file OUTPUT (line 10) and specifies the report-name EXPENSES (line 29). In addition to the RD entry and the control heading and control footing report groups previously discussed in this section, the Report Section describes a report heading (lines 40 through 42), a page heading (lines 43 through 49), a detail line (lines 63 through 67), and a page footing (lines 106 through 108).

The first time the GENERATE statement (line 118) is executed, seven report groups are processed and written on the report file: report heading, page heading, FINAL control heading, DEPT control heading, SECT control heading, ACCOUNT control heading, and detail. As input cards are read, detail lines are generated until a control break is detected. Control footing and control heading report groups are then processed according to the level at which the control break occurs. The control report groups are processed as follows:

Control break occurs in the ACCOUNT data item

- ACCOUNT control footing
- ACCOUNT control heading

Control break occurs in the SECT data item

- ACCOUNT control footing
- SECT control footing
- SECT control heading
- ACCOUNT control heading

Control break occurs in the DEPT data item

- ACCOUNT control footing
- SECT control footing
- DEPT control footing
- DEPT control heading
- SECT control heading
- ACCOUNT control heading

When the end of the input file is encountered, the TERMINATE statement (line 121) is executed. The control footing report groups are processed as if a control break occurred in the DEPT data item. The FINAL control footing report group, which prints the totals for all departments, is then processed. The page footing is the last report group written on the report file.

Figure 4-13 illustrates the input data used to create the EXPENSES report shown in figure 4-14.

```
1   IDENTIFICATION DIVISION.
2   PROGRAM-ID.  TERRTRY.
3   ENVIRONMENT DIVISION.
4   CONFIGURATION SECTION.
5   SOURCE-COMPUTER.  CYBER-170.
6   OBJECT-COMPUTER.  CYBER-170.
7   INPUT-OUTPUT SECTION.
8   FILE-CONTROL.
9       SELECT IN-FILE  ASSIGN TO ≠INPUT≠.
10      SELECT OUT-FILE ASSIGN TO ≠OUTPUT≠.
11  DATA DIVISION.
12  FILE SECTION.
13  FD  IN-FILE
14          LABEL RECORDS ARE OMITTED
15          DATA RECORD IS IN-REC.
16  01  IN-REC.
17      03  SALESMAN              PICTURE 9(4).
18      03  FILLER                PICTURE XXX.
19      03  TERRITORY             PICTURE XX.
20      03  FILLER                PICTURE XXX.
21      03  THIS-MONTH            PICTURE 9(5)V99.
22      03  FILLER                PICTURE XXX.
23      03  LAST-MONTH            PICTURE 9(5)V99.
24      03  FILLER                PICTURE XXX.
25      03  YTD                   PICTURE 9(6)V99.
26      03  FILLER                PICTURE X(40).
27  FD  OUT-FILE
28          LABEL RECORDS ARE OMITTED
29          REPORT IS TERRITORIES.
30  WORKING-STORAGE SECTION.
31  77  DATE-IN                   PICTURE X(14).
32  REPORT SECTION.
33  RD  TERRITORIES
34          CONTROLS ARE FINAL, TERRITORY
35          PAGE LIMIT IS 55 LINES
36              HEADING       3
37              FIRST DETAIL  7
38              LAST DETAIL  47
39              FOOTING      50.
40  01  TYPE IS PAGE HEADING.
41      03  LINE NUMBER 3  COLUMN NUMBER 56  PICTURE X(25)
42              VALUE IS ≠SALES REPORT BY TERRITORY≠.
43      03  LINE NUMBER PLUS 1  COLUMN NUMBER 62  PICTURE X(14)
44              SOURCE IS DATE-IN.
45  01  TYPE IS CONTROL HEADING TERRITORY  NEXT GROUP PLUS 1.
46      03  LINE NUMBER PLUS 1.
47          05  COLUMN NUMBER 23  PICTURE X(5)  VALUE IS ≠SALES≠.
48          05  COLUMN NUMBER 40  PICTURE X(8)  VALUE IS ≠SALESMAN≠.
49          05  COLUMN NUMBER 61  PICTURE X(4)  VALUE IS ≠LAST≠.
50          05  COLUMN NUMBER 85  PICTURE X(4)  VALUE IS ≠THIS≠.
51          05  COLUMN NUMBER 110 PICTURE XXX   VALUE IS ≠YTD≠.
52      03  LINE NUMBER PLUS 1.
53          05  COLUMN NUMBER 21  PICTURE X(9)  VALUE IS ≠TERRITORY≠.
54          05  COLUMN NUMBER 41  PICTURE X(6)  VALUE IS ≠NUMBER≠.
55          05  COLUMN NUMBER 61  PICTURE X(5)  VALUE IS ≠MONTH≠.
56          05  COLUMN NUMBER 85  PICTURE X(5)  VALUE IS ≠MONTH≠.
57          05  COLUMN NUMBER 109 PICTURE X(5)  VALUE IS ≠SALES≠.
58  01  DET-LINE  TYPE IS DETAIL  LINE NUMBER PLUS 1.
59      03  COLUMN NUMBER 24   PICTURE XX    SOURCE IS TERRITORY
60              GROUP INDICATE.
61      03  COLUMN NUMBER 42   PICTURE XXXX  SOURCE IS SALESMAN.
62      03  COLUMN NUMBER 58   PICTURE $$$,999.99
63              SOURCE IS LAST-MONTH.
64      03  COLUMN NUMBER 82   PICTURE $$$,999.99
65              SOURCE IS THIS-MONTH.
66      03  COLUMN NUMBER 106  PICTURE $$$$,999.99
67              SOURCE IS YTD.
68  01  TYPE IS CONTROL FOOTING TERRITORY  NEXT GROUP PLUS 6.
69      03  LINE NUMBER PLUS 1.
70          05  COLUMN NUMBER 57  PICTURE X(60)  VALUE IS ALL ≠-≠.
71      03  LINE NUMBER PLUS 1.
72          05  COLUMN NUMBER 21   PICTURE X(26)
```

Figure 4-9. TERRTRY Program (Sheet 1 of 2)

```
                    VALUE IS ≠TOTALS FOR SALES TERRITORY≠.
               05  COLUMN NUMBER 48  PICTURE XX  SOURCE IS TERRITORY.
               05  TOT-1  COLUMN NUMBER 57   PICTURE $$$$,999.99
                    SUM LAST-MONTH.
               05  TOT-2  COLUMN NUMBER 81   PICTURE $$$$,999.99
                    SUM THIS-MONTH.
               05  TOT-3  COLUMN NUMBER 104  PICTURE $$,$$$,999.99
                    SUM YTD.
     01  TYPE IS CONTROL FOOTING FINAL.
         03  LINE NUMBER PLUS 2.
             05  COLUMN NUMBER 55   PICTURE X(62)  VALUE IS ALL ≠*≠.
         03  LINE NUMBER PLUS 2.
             05  COLUMN NUMBER 21   PICTURE X(32)
                    VALUE IS ≠TOTALS FOR ALL SALES TERRITORIES≠.
             05  COLUMN NUMBER 55   PICTURE $$,$$$,999.99  SUM TOT-1.
             05  COLUMN NUMBER 79   PICTURE $$,$$$,999.99  SUM TOT-2.
             05  COLUMN NUMBER 103  PICTURE $$$,$$$,999.99 SUM TOT-3.
     01  TYPE IS PAGE FOOTING  LINE NUMBER 54.
         03  COLUMN NUMBER 21   PICTURE X(25)
                VALUE IS ≠SALES REPORT BY TERRITORY≠.
         03  COLUMN NUMBER 109  PICTURE X(4)  VALUE IS ≠PAGE≠.
         03  COLUMN NUMBER 114  PICTURE ZZ9   SOURCE IS PAGE-COUNTER.
    PROCEDURE DIVISION.
    OPENING.
         OPEN INPUT IN-FILE.
         OPEN OUTPUT OUT-FILE.
         ACCEPT DATE-IN.
         INITIATE TERRITORIES.
    READING.
         READ IN-FILE RECORD
             AT END GO TO CLOSING.
         GENERATE DET-LINE.
         GO TO READING.
    CLOSING.
         TERMINATE TERRITORIES.
         CLOSE IN-FILE, OUT-FILE.
         STOP RUN.
```

Figure 4-9. TERRTRY Program (Sheet 2 of 2)



Figure 4-10. Input Data for TERRTRY Program

SALES REPORT BY TERRITORY
JUNE 1976

SALES TERRITORY	SALESMAN NUMBER	LAST MONTH	THIS MONTH	YTD SALES
NE	1436	$13,569.87	$14,772.09	$65,310.48
	4280	$15,373.65	$16,043.77	$72,920.15
	5772	$12,904.18	$15,254.83	$68,240.38
	7244	$9,910.34	$12,878.60	$55,791.25
TOTALS FOR SALES TERRITORY NE		$51,758.04	$58,949.29	$262,262.26

SALES TERRITORY	SALESMAN NUMBER	LAST MONTH	THIS MONTH	YTD SALES
NO	3124	$9,250.61	$9,845.06	$48,341.03
	5506	$10,876.26	$13,016.95	$63,185.10
	6578	$11,490.75	$10,768.23	$58,922.68
	8321	$12,870.00	$13,105.23	$69,132.54
TOTALS FOR SALES TERRITORY NO		$44,487.62	$46,735.47	$239,581.35

SALES TERRITORY	SALESMAN NUMBER	LAST MONTH	THIS MONTH	YTD SALES
NW	4013	$12,456.22	$12,165.90	$61,443.78
	6311	$8,538.62	$7,912.40	$44,651.27
	9787	$11,457.20	$9,910.78	$57,938.10
TOTALS FOR SALES TERRITORY NW		$32,452.04	$29,989.08	$164,033.15

SALES REPORT BY TERRITORY PAGE 1

Figure 4-11. TERRITORIES Report Generated by TERRTRY Program (Sheet 1 of 2)

SALES REPORT BY TERRITORY
JUNE 1976

SALES TERRITORY	SALESMAN NUMBER	LAST MONTH	THIS MONTH	YTD SALES
SE	2275	$9,448.20	$11,295.16	$52,489.96
	3865	$11,298.45	$8,933.42	$50,617.64
	7505	$13,678.03	$14,029.30	$67,904.34
	9178	$14,806.83	$16,852.43	$79,243.55
TOTALS FOR SALES TERRITORY SE		$49,231.51	$51,110.31	$250,255.49

SALES TERRITORY	SALESMAN NUMBER	LAST MONTH	THIS MONTH	YTD SALES
SO	1899	$6,225.14	$8,321.45	$38,762.25
	4949	$9,983.51	$9,568.79	$48,056.33
	8626	$9,562.41	$10,429.16	$48,061.26
TOTALS FOR SALES TERRITORY SO		$25,771.06	$28,319.40	$134,879.84

SALES TERRITORY	SALESMAN NUMBER	LAST MONTH	THIS MONTH	YTD SALES
SW	1062	$11,457.93	$9,465.38	$49,665.15
	2361	$10,623.54	$10,454.88	$51,124.59
	7092	$9,987.60	$9,354.52	$47,364.43
	7890	$10,393.66	$8,896.45	$45,887.20
TOTALS FOR SALES TERRITORY SW		$42,462.73	$38,171.23	$194,041.37
TOTALS FOR ALL SALES TERRITORIES		$246,163.00	$253,274.78	$1,245,053.46

SALES REPORT BY TERRITORY PAGE 2

Figure 4-11. TERRITORIES Report Generated by TERRTRY Program (Sheet 2 of 2)

```
 1   IDENTIFICATION DIVISION.
 2   PROGRAM-ID.  BUDGETS.
 3   ENVIRONMENT DIVISION.
 4   CONFIGURATION SECTION.
 5   SOURCE-COMPUTER.  CYBER-170.
 6   OBJECT-COMPUTER.  CYBER-170.
 7   INPUT-OUTPUT SECTION.
 8   FILE-CONTROL.
 9       SELECT CARD-IN ASSIGN TO ≠INPUT≠.
10       SELECT PRINT-FILE ASSIGN TO ≠OUTPUT≠.
11   DATA DIVISION.
12   FILE SECTION.
13   FD  CARD-IN
14           LABEL RECORDS ARE OMITTED
15           DATA RECORD IS  CARD-REC.
16   01  CARD-REC.
17       03  DEPT                PICTURE X(5).
18       03  FILLER              PICTURE XXX.
19       03  SECT                PICTURE X(4).
20       03  FILLER              PICTURE XXX.
21       03  ACCOUNT             PICTURE 9(5).
22       03  FILLER              PICTURE XXX.
23       03  BUDGET              PICTURE 9(4)V99.
24       03  FILLER              PICTURE XXX.
25       03  ACTUAL              PICTURE 9(4)V99.
26       03  FILLER              PICTURE X(42).
27   FD  PRINT-FILE
28           LABEL RECORDS ARE OMITTED
29           REPORT IS EXPENSES.
30   WORKING-STORAGE SECTION.
31   77  MONTH-IN                PICTURE 99.
32   REPORT SECTION.
33   RD  EXPENSES
34           CONTROLS ARE FINAL, DEPT, SECT, ACCOUNT
35           PAGE LIMIT IS 55 LINES
36               HEADING       2
37               FIRST DETAIL  7
38               LAST DETAIL   50
39               FOOTING       52.
40   01  TYPE IS REPORT HEADING  LINE NUMBER 2.
41       03  COLUMN NUMBER 43  PICTURE X(52)  VALUE IS ≠B U D G E T
42   -           ≠ V S    A C T U A L    E X P E N S E S≠.
43   01  TYPE IS PAGE HEADING.
44       03  LINE NUMBER PLUS 3.
45           05  COLUMN NUMBER 61  PICTURE X(6)  VALUE IS ≠BUDGET≠.
46           05  COLUMN NUMBER 101 PICTURE X(6)  VALUE IS ≠ACTUAL≠.
47       03  LINE NUMBER PLUS 1.
48           05  COLUMN NUMBER 49  PICTURE X(30) VALUE IS ALL ≠-≠.
49           05  COLUMN NUMBER 89  PICTURE X(30) VALUE IS ALL ≠-≠.
50   01  TYPE IS CONTROL HEADING FINAL  LINE NUMBER PLUS 2.
51       03  COLUMN NUMBER 19  PICTURE X(16)
52               VALUE IS ≠***REPORT COVERS≠.
53       03  COLUMN NUMBER 36  PICTURE Z9    SOURCE IS MONTH-IN.
54       03  COLUMN NUMBER 39  PICTURE X(9)  VALUE IS ≠MONTHS***≠.
```

Figure 4-12. BUDGETS Program (Sheet 1 of 3)

```
01  TYPE IS CONTROL HEADING DEPT  LINE NUMBER PLUS 3.
    03  COLUMN NUMBER 22  PICTURE X(10)  VALUE IS ≠DEPARTMENT≠.
    03  COLUMN NUMBER 33  PICTURE X(5)   SOURCE IS DEPT.
01  TYPE IS CONTROL HEADING SECT  LINE NUMBER PLUS 2.
    03  COLUMN NUMBER 25  PICTURE X(7)   VALUE IS ≠SECTION≠.
    03  COLUMN NUMBER 33  PICTURE X(4)   SOURCE IS SECT.
01  TYPE IS CONTROL HEADING ACCOUNT  LINE NUMBER PLUS 2.
    03  COLUMN NUMBER 33  PICTURE X(7)   VALUE IS ≠ACCOUNT≠.
01  DETAIL-LINE  TYPE IS DETAIL  LINE NUMBER PLUS 1.
    03  COLUMN NUMBER 34  PICTURE X(5)   SOURCE IS ACCOUNT
            GROUP INDICATE.
    03  COLUMN NUMBER 50  PICTURE $$,999.99  SOURCE IS BUDGET.
    03  COLUMN NUMBER 90  PICTURE $$,999.99  SOURCE IS ACTUAL.
01  TYPE IS CONTROL FOOTING ACCOUNT.
    03  LINE NUMBER PLUS 1.
        05  COLUMN NUMBER 49  PICTURE X(10)  VALUE IS ALL ≠-≠.
        05  COLUMN NUMBER 89  PICTURE X(10)  VALUE IS ALL ≠-≠.
    03  LINE NUMBER PLUS 1.
        05  BUD-1  COLUMN NUMBER 49  PICTURE $$$,999.99
                SUM BUDGET.
        05  ACT-1  COLUMN NUMBER 89  PICTURE $$$,999.99
                SUM ACTUAL.
    03  LINE NUMBER PLUS 1.
        05  COLUMN NUMBER 25  PICTURE X(22)
                VALUE IS ≠ACCUM TOTALS - SECTION≠.
        05  BUD-2  COLUMN NUMBER 68   PICTURE $$$$,999.99
                SUM BUD-1 RESET ON SECT.
        05  ACT-2  COLUMN NUMBER 108  PICTURE $$$$,999.99
                SUM ACT-1 RESET ON SECT.
01  TYPE IS CONTROL FOOTING SECT  LINE NUMBER PLUS 2.
    03  COLUMN NUMBER 22   PICTURE X(25)
            VALUE IS ≠ACCUM TOTALS - DEPARTMENT≠.
    03  COLUMN NUMBER 66   PICTURE $$,$$$,999.99
            SUM BUD-1 RESET ON DEPT.
    03  COLUMN NUMBER 106  PICTURE $$,$$$,999.99
            SUM ACT-1 RESET ON DEPT.
01  TYPE IS CONTROL FOOTING DEPT  LINE NUMBER PLUS 2
        NEXT GROUP NEXT PAGE.
    03  COLUMN NUMBER 19  PICTURE X(100)  VALUE ALL ≠*≠.
01  TYPE IS CONTROL FOOTING FINAL.
    03  LINE NUMBER PLUS 4.
        05  COLUMN NUMBER 16   PICTURE X(106)  VALUE IS ALL ≠*≠.
    03  LINE NUMBER PLUS 1.
        05  COLUMN NUMBER 19   PICTURE X(30)
                VALUE IS ≠GRAND TOTALS - ALL DEPARTMENTS≠.
        05  COLUMN NUMBER 65   PICTURE $$$,$$$,999.99
                SUM BUD-1 RESET ON FINAL.
        05  COLUMN NUMBER 105  PICTURE $$$,$$$,999.99
                SUM ACT-1 RESET ON FINAL.
    03  LINE NUMBER PLUS 1.
        05  COLUMN NUMBER 16   PICTURE X(106)  VALUE IS ALL ≠*≠.
01  TYPE IS PAGE FOOTING  LINE NUMBER 55.
    03  COLUMN NUMBER 114  PICTURE X(4)  VALUE IS ≠PAGE≠.
    03  COLUMN NUMBER 119  PICTURE ZZ9   SOURCE IS PAGE-COUNTER.
```

Figure 4-12. BUDGETS Program (Sheet 2 of 3)

```
109   PROCEDURE DIVISION.
110   INITIALIZATION.
111       OPEN INPUT CARD-IN.
112       OPEN OUTPUT PRINT-FILE.
113       ACCEPT MONTH-IN.
114       INITIATE EXPENSES.
115   READ-CARD.
116       READ CARD-IN RECORD
117           AT END GO TO TERMINATION.
118       GENERATE DETAIL-LINE.
119       GO TO READ-CARD.
120   TERMINATION.
121       TERMINATE EXPENSES.
122       CLOSE CARD-IN, PRINT-FILE.
123       STOP RUN.
```

Figure 4-12. BUDGETS Program (Sheet 3 of 3)

```
                  06
Column 1          Column 9  Column 16  Column 24  Column 33
PRSNL     PS10    24689     102500     114650
PRSNL     PS10    24689     085000     067500
PRSNL     PS10    24689     243800     238160
PRSNL     PS10    48153     505000     519275
PRSNL     PS30    24689     116500     094150
PRSNL     PS30    24689     305000     332525
PRSNL     PS30    48153     813500     806975
PRGMG     PG25    51960     648500     612800
PRGMG     PG25    51960     284000     356045
PRGMG     PG25    87013     056500     041580
PRGMG     PG25    87013     124900     116250
PRGMG     PG80    51960     450000     518575
PRGMG     PG80    87013     739500     772125
PRGMG     PG80    87013     541600     581050
ACCTG     AC45    10495     362500     332190
ACCTG     AC45    62377     601600     615570
ACCTG     AC45    62377     495400     493200
ACCTG     AC45    62377     326500     331585
ACCTG     AC70    10495     518600     452910
ACCTG     AC70    10495     247500     215625
ACCTG     AC70    62377     664000     653550
```

Figure 4-13. Input Data for BUDGETS Program

B U D G E T V S A C T U A L E X P E N S E S

	BUDGET		ACTUAL	
REPORT COVERS 6 MONTHS				
DEPARTMENT PRSNL				
SECTION PS10				
ACCOUNT 24689	$1,025.00		$1,146.50	
	$850.00		$675.00	
	$2,438.00		$2,381.60	
	$4,313.00		$4,203.10	
ACCUM TOTALS - SECTION		$4,313.00		$4,203.10
ACCOUNT 48153	$5,050.00		$5,192.75	
	$5,050.00		$5,192.75	
ACCUM TOTALS - SECTION		$9,363.00		$9,395.85
ACCUM TOTALS - DEPARTMENT		$9,363.00		$9,395.85
SECTION PS30				
ACCOUNT 24689	$1,165.00		$941.50	
	$3,050.00		$3,325.25	
	$4,215.00		$4,266.75	
ACCUM TOTALS - SECTION		$4,215.00		$4,266.75
ACCOUNT 48153	$8,135.00		$8,069.75	
	$8,135.00		$8,069.75	
ACCUM TOTALS - SECTION		$12,350.00		$12,336.50
ACCUM TOTALS - DEPARTMENT		$21,713.00		$21,732.35

****

PAGE 1

Figure 4-14. EXPENSES Report Generated by BUDGETS Program (Sheet 1 of 3)

	BUDGET		ACTUAL	
DEPARTMENT PRGMG				
SECTION PG25				
ACCOUNT 51960	$6,485.00		$6,128.00	
	$2,840.00		$3,560.45	
	$9,325.00		$9,688.45	
ACCUM TOTALS - SECTION		$9,325.00		$9,688.45
ACCOUNT 87013	$565.00		$415.80	
	$1,249.00		$1,162.50	
	$1,814.00		$1,578.30	
ACCUM TOTALS - SECTION		$11,139.00		$11,266.75
ACCUM TOTALS - DEPARTMENT		$11,139.00		$11,266.75
SECTION PG80				
ACCOUNT 51960	$4,500.00		$5,185.75	
	$4,500.00		$5,185.75	
ACCUM TOTALS - SECTION		$4,500.00		$5,185.75
ACCOUNT 87013	$7,395.00		$7,721.25	
	$5,416.00		$5,810.50	
	$12,811.00		$13,531.75	
ACCUM TOTALS - SECTION		$17,311.00		$18,717.50
ACCUM TOTALS - DEPARTMENT		$28,450.00		$29,984.25

****

PAGE 2

Figure 4-14. EXPENSES Report Generated by BUDGETS Program (Sheet 2 of 3)

	BUDGET		ACTUAL	
DEPARTMENT ACCTG				
SECTION AC45				
ACCOUNT 10495	$3,625.00		$3,321.90	
	$3,625.00		$3,321.90	
ACCUM TOTALS - SECTION		$3,625.00		$3,321.90
ACCOUNT 62377	$6,016.00		$6,155.70	
	$4,954.00		$4,932.00	
	$3,265.00		$3,315.85	
	$14,235.00		$14,403.55	
ACCUM TOTALS - SECTION		$17,860.00		$17,725.45
ACCUM TOTALS - DEPARTMENT		$17,860.00		$17,725.45
SECTION AC70				
ACCOUNT 10495	$5,186.00		$4,529.10	
	$2,475.00		$2,156.25	
	$7,661.00		$6,685.35	
ACCUM TOTALS - SECTION		$7,661.00		$6,685.35
ACCOUNT 62377	$6,640.00		$6,535.50	
	$6,640.00		$6,535.50	
ACCUM TOTALS - SECTION		$14,301.00		$13,220.85
ACCUM TOTALS - DEPARTMENT		$32,161.00		$30,946.30
GRAND TOTALS - ALL DEPARTMENTS		$82,324.00		$82,662.90

PAGE 3

Figure 4-14. EXPENSES Report Generated by BUDGETS Program (Sheet 3 of 3)

At times it is desirable to generate more than one report from the information in the input records. The input file can be read once and entries for the various reports can be written on the report file as each input record is read. At program termination, the report file contains the entries for all the reports in the order in which the entries were generated. The entries can then be separated by a user program based on codes that identify the entries for specific reports. The File Organization and Record Manager (FORM) utility program can also be used to separate the entries and output the reports to the printer.

The sample program presented in this section generates two reports from the input records. The SUMMARY report provides a listing of invoice totals by customer. The INVOICES report lists detailed information related to a customer's invoices.

DESIGNATING THE REPORT FILE

The report file is specified in the SELECT clause of the Environment Division and in the File Description entry of the Data Division. The SELECT clause includes the file-name used by the program and the logical file name used by the system. Because the entries for the reports are intermixed on the report file and must be separated before printing, the report file is not assigned to the system file OUTPUT.

The File Description entry specifies the program file-name and the report-name of each report to be generated. The LABEL RECORDS clause must also be included in the File Description entry. No other clauses can be specified.

Figure 5-1 shows the File Description entry for the INV-RPT program at the end of this section. The two reports generated by the program are named in the REPORTS clause. The SELECT clause, which is also shown in figure 5-1, assigns the logical file name RPTFLE to the report file. The report file is a sequential disk file that must be preserved at the end of the job for later processing.

DESCRIBING EACH REPORT

Each report named by the REPORTS clause in the File Description entry is described in the Report Section by a separate Report Description entry. This entry includes the RD entry and any of the Report Group Description entries discussed in sections 3 and 4.

```
ENVIRONMENT DIVISION.
     :
     :
     SELECT RPT-FILE ASSIGN TO RPTFLE.
     :
     :
DATA DIVISION.
FILE SECTION.
     :
     :
FD   RPT-FLE
          LABEL RECORDS ARE OMITTED
          REPORTS ARE INVOICES, SUMMARY.
```

Figure 5-1. File Description Entry for INV-RPT Program

The RD entry contains an additional clause when multiple reports are being generated. The CODE clause specifies a two-character nonnumeric literal that is used to identify the entries for the report. When an entry is generated for the report, Report Writer adds the specified identifier as the first two characters of the record written on the report file. Records are then 139 characters in length: the two-character identifier, the carriage control character, and the 136 characters of the print line. When the report file is processed for printing, the identifiers can be used to separate the entries for the various reports.

When the CODE clause is specified in the RD entry for one report, it must be specified in the RD entry for all reports named in the same REPORTS clause.

The RD entries for the INV-RPT program are shown in figure 5-2. Both entries include the CODE clause. The identifier CS is prefixed to the entries for the SUMMARY report and the identifier CI is prefixed to the entries for the INVOICES report.

```
RD   SUMMARY
          CODE "CS"
          CONTROLS ARE FINAL, CUST-NAME
          PAGE LIMITS ARE 50 LINES
               HEADING       2
               FIRST DETAIL  9
               LAST DETAIL   44
               FOOTING       46.
     :
     :
RD   INVOICES
          CODE "CI"
          CONTROLS ARE CUST-NAME, INVOICE-NO
          PAGE LIMIT IS 50 LINES
               HEADING       2
               FIRST DETAIL  9
               LAST DETAIL   44
               FOOTING       46.
```

Figure 5-2. RD Entries for INV-RPT Program

PRINTING THE REPORTS

At program termination, the report file contains the entries generated for all the reports. These entries must be separated before being output to the printer. The report file, which must be preserved as a permanent file when the report generation program is terminated, can be processed by the FORM utility program or by a user program that separates the report entries on the basis of the identifier codes associated with the entries.

The report file records are 139 characters in length. The first two characters of a record are the code that identifies the report to which the entry belongs. The remaining characters are the carriage control character and the 136

characters of the print line. The two-character code field is used only to separate the entries. The rest of the record is output to the printer.

INV-RPT PROGRAM

The INV-RPT program generates two reports: SUMMARY and INVOICES. The report worksheets for these two reports are shown in figures 5-3 and 5-4. The program that generates the reports is shown in figure 5-5.

The input data for these reports is contained in two different types of records. Records with code letter A in the first character position contain information pertaining to the complete invoice for a customer. Records with code letter B in the first character position contain information pertaining to individual items in the invoice. The SUMMARY report uses data from the code A record only; the INVOICES report uses data from both types of records.

The SUMMARY report processes two control breaks. When a detail line is being generated and a control break occurs in the CUST-NAME data item, the CUST-NAME control footing report group (lines 70 through 74) is written on the report file. This report group includes a sum counter that produces the total amount of all invoices for a customer. The major control break, FINAL, is processed at report termination. The FINAL control footing report group (lines 75 through 78) is the last report group written on the report file; it includes a sum counter that produces the total amount for all invoices on the report.

The INVOICES report has two break control items. A control heading report group is described for the major control break CUST-NAME (lines 100 through 102) and a control footing report group is described for the minor control break INVOICE-NO (lines 109 through 114). The control heading report group produces a line containing the name of the customer. The control footing report group includes a sum counter that totals the amount for an invoice.

The first input record for each invoice is a code A record, which is then followed by one code B record for each item on the invoice. When a code A record is read (line 135), the GENERATE statement for the SUMMARY report is executed (line 131). Each time a code B record is read, the GENERATE statement for the INVOICES report is executed (line 138). Figure 5-6 illustrates the input records for the INV RPT program. When the program terminates, the report file is preserved for subsequent processing.

The FORM directives shown in figure 5-7 can be used to print the two reports. Two FORM passes are required to output both reports. The following directives are specified for each pass:

INP	Designates the logical file name of the report file as the input file.
OUT	Designates the system file OUTPUT as the output file and specifies the backgrounding of the output record is blanks.
QAL	Qualifies the input records for output. For the first pass, only those records with the characters CS in the first two positions qualify for output. For the second pass, the characters CI in the first two positions qualify the record for output.
REF	Reformats the input record for output. Beginning in position 1 of the output record, 137 characters are moved from the input record starting with character 3. This deletes the two-character code from the report file record.
PRT	Specifies page formatting for the output. FMT=A indicates that the first character of the output record is a carriage control character.
XEQ	Indicates the end of directives for a single pass. XEQ(FIN) indicates the end of the final pass.

Refer to the FORM Reference Manual for a complete description of the FORM directives.

The program shown in figure 5-8 can also be executed to separate the entries for the two reports and to output the reports to the line printer. The SUMMARY report is shown in figure 5-9 and the INVOICES report is shown in figure 5-10.



Figure 5-3. Report Worksheet for SUMMARY Report



Figure 5-4. Report Worksheet for INVOICES Report

```
 1   IDENTIFICATION DIVISION.
 2   PROGRAM-ID.  INV-RPT.
 3   ENVIRONMENT DIVISION.
 4   CONFIGURATION SECTION.
 5   SOURCE-COMPUTER.  CYBER-170.
 6   OBJECT-COMPUTER.  CYBER-170.
 7   INPUT-OUTPUT SECTION.
 8   FILE-CONTROL.
 9       SELECT CARD-FILE ASSIGN TO ≠INPUT≠.
10       SELECT RPT-FILE  ASSIGN TO RPTFLE.
11   DATA DIVISION.
12   FILE SECTION.
13   FD  CARD-FILE
14           LABEL RECORDS ARE OMITTED
15           DATA RECORDS ARE REC-1, REC-2.
16   01  REC-1.
17       03  CODE-1                  PICTURE X.
18       03  C-NAME                  PICTURE X(20).
19       03  INV-NO                  PICTURE 9(6).
20       03  FILLER                  PICTURE XX.
21       03  INV-AMT                 PICTURE 9(5)V99.
22       03  FILLER                  PICTURE X(44).
23   01  REC-2.
24       03  CODE-2                  PICTURE X.
25       03  FILLER                  PICTURE X(8).
26       03  ITEM-NO                 PICTURE 999.
27       03  FILLER                  PICTURE XX.
28       03  QUANTITY                PICTURE 999.
29       03  FILLER                  PICTURE XX.
30       03  COST-PER-UNIT           PICTURE 99V99.
31       03  FILLER                  PICTURE X(57).
32   FD  RPT-FILE
33           LABEL RECORDS ARE OMITTED
34           REPORTS ARE INVOICES, SUMMARY.
35   WORKING-STORAGE SECTION.
36   77  DATE-IN                     PICTURE X(8).
37   77  ITEM-AMT                    PICTURE 9(4)V99.
38   77  CUST-NAME                   PICTURE X(20).
39   77  INVOICE-NO                  PICTURE 9(6).
40   REPORT SECTION.
41   RD  SUMMARY
42           CODE ≠CS≠
43           CONTROLS ARE FINAL, CUST-NAME
44           PAGE LIMITS ARE 50 LINES
45               HEADING      2
46               FIRST DETAIL 9
47               LAST DETAIL 44
48               FOOTING     46.
49   01  TYPE IS PAGE HEADING.
50       03  LINE NUMBER 2.
51           05  COLUMN NUMBER 51  PICTURE X(18)
52                   VALUE IS ≠CUSTOMER SUMMARY -≠.
53           05  COLUMN NUMBER 70  PICTURE X(8)  SOURCE IS DATE-IN.
54           05  COLUMN NUMBER 100 PICTURE X(4)  VALUE IS ≠PAGE≠.
55           05  COLUMN NUMBER 105 PICTURE ZZ9
56                   SOURCE IS PAGE-COUNTER.
57       03  LINE NUMBER PLUS 3.
58           05  COLUMN NUMBER 77  PICTURE X(7)  VALUE IS ≠INVOICE≠.
59       03  LINE NUMBER PLUS 1.
60           05  COLUMN NUMBER 33  PICTURE X(13)
61                   VALUE IS ≠CUSTOMER NAME≠.
62           05  COLUMN NUMBER 77  PICTURE X(6)  VALUE IS ≠NUMBER≠.
63           05  COLUMN NUMBER 101 PICTURE X(6)  VALUE IS ≠AMOUNT≠.
64   01  DET-LINE  TYPE IS DETAIL  LINE NUMBER PLUS 1.
65       03  COLUMN NUMBER 30  PICTURE X(20)  SOURCE IS CUST-NAME
66               GROUP INDICATE.
67       03  COLUMN NUMBER 77  PICTURE 9(6)   SOURCE IS INVOICE-NO.
68       03  COLUMN NUMBER 99  PICTURE ZZ,ZZ9.99
69               SOURCE IS INV-AMT.
70   01  TYPE IS CONTROL FOOTING CUST-NAME  LINE NUMBER PLUS 2
71           NEXT GROUP PLUS 2.
72       03  COLUMN NUMBER 38  PICTURE X(9)  VALUE IS ≠TOTAL FOR≠.
73       03  COLUMN NUMBER 48  PICTURE X(20) SOURCE IS CUST-NAME.
74       03  COLUMN NUMBER 97  PICTURE $$$$,$$9.99  SUM INV-AMT.
```

Figure 5-5. INV-RPT Program (Sheet 1 of 2)

```
01  TYPE IS CONTROL FOOTING FINAL  LINE NUMBER PLUS 4.
    03  COLUMN NUMBER 38  PICTURE X(24)
            VALUE IS ≠TOTAL OF ALL INVOICES IS≠.
    03  COLUMN NUMBER 63  PICTURE $$$$,$$9.99  SUM INV-AMT.
RD  INVOICES
        CODE ≠CI≠
        CONTROLS ARE CUST-NAME, INVOICE-NO
        PAGE LIMIT IS 50 LINES
            HEADING      2
            FIRST DETAIL 9
            LAST DETAIL 44
            FOOTING     46.
01  TYPE IS REPORT HEADING  LINE NUMBER 2.
    03  COLUMN NUMBER 50  PICTURE X(33)
            VALUE IS ≠C U S T O M E R   I N V O I C E S≠.
    03  COLUMN NUMBER 103 PICTURE X(8)  SOURCE IS DATE-IN.
01  TYPE IS PAGE HEADING.
    03  LINE NUMBER PLUS 3.
        05  COLUMN NUMBER 27  PICTURE X(7)  VALUE IS ≠INVOICE≠.
        05  COLUMN NUMBER 49  PICTURE X(4)  VALUE IS ≠ITEM≠.
    03  LINE NUMBER PLUS 1.
        05  COLUMN NUMBER 27  PICTURE X(6)  VALUE IS ≠NUMBER≠.
        05  COLUMN NUMBER 48  PICTURE X(6)  VALUE IS ≠NUMBER≠.
        05  COLUMN NUMBER 68  PICTURE X(8)  VALUE IS ≠QUANTITY≠.
        05  COLUMN NUMBER 93  PICTURE X(6)  VALUE IS ≠AMOUNT≠.
01  TYPE IS CONTROL HEADING CUST-NAME  LINE NUMBER PLUS 3
        NEXT GROUP PLUS 1.
    03  COLUMN NUMBER 27  PICTURE X(20)  SOURCE IS CUST-NAME.
01  LINE-ITEM  TYPE IS DETAIL  LINE NUMBER PLUS 1.
    03  COLUMN NUMBER 27  PICTURE 9(6)       SOURCE IS INVOICE-NO
            GROUP INDICATE.
    03  COLUMN NUMBER 49  PICTURE 999        SOURCE IS ITEM-NO.
    03  COLUMN NUMBER 70  PICTURE ZZ9        SOURCE IS QUANTITY.
    03  COLUMN NUMBER 92  PICTURE Z,ZZ9.99   SOURCE IS ITEM-AMT.
01  TYPE IS CONTROL FOOTING INVOICE-NO  LINE NUMBER PLUS 2
        NEXT GROUP PLUS 1.
    03  COLUMN NUMBER 64  PICTURE X(13)
            VALUE IS ≠INVOICE TOTAL≠.
    03  SUM-1  COLUMN NUMBER 90  PICTURE $$$,$$9.99
            SUM ITEM-AMT.
01  TYPE IS PAGE FOOTING  LINE NUMBER 50.
    03  COLUMN NUMBER 103 PICTURE X(4)  VALUE IS ≠PAGE≠.
    03  COLUMN NUMBER 108 PICTURE ZZ9   SOURCE IS PAGE-COUNTER.
PROCEDURE DIVISION.
OPENING.
    OPEN INPUT CARD-FILE.
    OPEN OUTPUT RPT-FILE.
    ACCEPT DATE-IN.
    INITIATE INVOICES, SUMMARY.
    READ CARD-FILE RECORD
        AT END GO TO ERR-1.
    IF CODE-1 NOT EQUAL TO ≠A≠
        GO TO ERR-1.
READ-A.
    MOVE C-NAME TO CUST-NAME.
    MOVE INV-NO TO INVOICE-NO.
    GENERATE DET-LINE.
READ-B.
    READ CARD-FILE RECORD
        AT END GO TO CLOSING.
    IF CODE-2 EQUALS ≠A≠ GO TO READ-A
        ELSE IF CODE-2 NOT EQUAL TO ≠B≠ GO TO ERR-1.
    COMPUTE ITEM-AMT = QUANTITY * COST-PER-UNIT.
    GENERATE LINE-ITEM.
    GO TO READ-B.
ERR-1.
    DISPLAY ≠BAD INPUT DECK≠.
    STOP RUN.
CLOSING.
    TERMINATE INVOICES, SUMMARY.
    CLOSE CARD-FILE, RPT-FILE.
    STOP RUN.
```

Figure 5-5. INV-RPT Program (Sheet 2 of 2)

```
06/15/76
  Column 1                    Column 22
                Column 10                 Column 30
ADOWNTOWN SALES CO    175256   0020365
B         465  005  1095
B         103  020  0495
B         916  002  2495
ADOWNTOWN SALES CO    180696   0101575
B         103  030  0495
B         456  045  1650
B         916  005  2495
AIDEAL  SALES INC      075258   0275000
B         309  100  1475
B         416  085  1500
ALOW COST SALES CO    138966   0125315
B         532  025  0575
B         901  060  1350
B         319  012  2495
ALOW COST SALES CO    149125   0035000
B         029  002  2500
B         112  030  1000
```

Figure 5-6. Input Data for INV-RPT Program

```
INP(RPTFLE)
OUT(OUTPUT,BGD=X)
QAL(OUTPUT,1X2 EQ $CS$)
REF(OUTPUT,1X=3X137)
PRT(OUTPUT,FMT=A)
XEQ.
INP(RPTFLE)
OUT(OUTPUT,BGD=X)
QAL(OUTPUT,1X2 EQ $CI$)
REF(OUTPUT,1X=3X137)
PRT(OUTPUT,FMT=A)
XEQ(FIN)
```

Figure 5-7. FORM Directives

```
 1   IDENTIFICATION DIVISION.
 2   PROGRAM-ID.  WRITERP.
 3   ENVIRONMENT DIVISION.
 4   CONFIGURATION SECTION.
 5   SOURCE-COMPUTER.  CYBER-170.
 6   OBJECT-COMPUTER.  CYBER-170.
 7   INPUT-OUTPUT SECTION.
 8   FILE-CONTROL.
 9       SELECT RPT-FILE ASSIGN TO RPTFLE
10           USE ≠RT=Z,BT=C≠.
11       SELECT PRINT-FILE ASSIGN TO ≠OUTPUT≠.
12   DATA DIVISION.
13   FILE SECTION.
14   FD  RPT-FILE
15           LABEL RECORDS ARE OMITTED
16           DATA RECORD IS REPORT-REC.
17   01  REPORT-REC.
18       03  RPT-CODE              PICTURE XX.
19       03  RPT-LINE              PICTURE X(137).
20   FD  PRINT-FILE
21           LABEL RECORDS ARE OMITTED
22           DATA RECORD IS PRINTLINE.
23   01  PRINTLINE                 PICTURE X(137).
24   WORKING-STORAGE SECTION.
25   77  TEMP                      PICTURE XX.
26   PROCEDURE DIVISION.
27   OPEN-FILES.
28       OPEN INPUT RPT-FILE.
29       OPEN OUTPUT PRINT-FILE.
30       MOVE ≠CI≠ TO TEMP.
31   READ-FILE.
32       READ RPT-FILE RECORD
33           AT END GO TO CLOSE-FILES.
34       IF RPT-CODE EQUALS TEMP
35           WRITE PRINTLINE FROM RPT-LINE.
36       GO TO READ-FILE.
37   CLOSE-FILES.
38       CLOSE RPT-FILE.
39       IF TEMP EQUALS ≠CS≠
40           CLOSE PRINT-FILE
41           STOP RUN.
42       MOVE ≠CS≠ TO TEMP.
43       OPEN INPUT RPT-FILE.
44       GO TO READ-FILE.
```

Figure 5-8. WRITERP Program

CUSTOMER SUMMARY - 06/15/76 PAGE 1

CUSTOMER NAME	INVOICE NUMBER	AMOUNT
DOWNTOWN SALES CO	175256	203.65
	180696	1,015.75
TOTAL FOR DOWNTOWN SALES CO		$1,219.40
IDEAL SALES INC	075258	2,750.00
TOTAL FOR IDEAL SALES INC		$2,750.00
LOW COST SALES CO	138966	1,253.15
	149125	350.00
TOTAL FOR LOW COST SALES CO		$1,603.15

TOTAL OF ALL INVOICES IS $5,572.55

Figure 5-9. SUMMARY Report Generated by INV-RPT Program

C U S T O M E R I N V O I C E S 06/15/76

INVOICE NUMBER	ITEM NUMBER	QUANTITY	AMOUNT
DOWNTOWN SALES CO			
175256	465	5	54.75
	103	20	99.00
	916	2	49.90
		INVOICE TOTAL	$203.65
180696	103	30	148.50
	456	45	742.50
	916	5	124.75
		INVOICE TOTAL	$1,015.75
IDEAL SALES INC			
075258	309	100	1,475.00
	416	85	1,275.00
		INVOICE TOTAL	$2,750.00
LOW COST SALES CO			
138966	532	25	143.75
	901	60	810.00
	319	12	299.40
		INVOICE TOTAL	$1,253.15
149125	029	2	50.00
	112	30	300.00
		INVOICE TOTAL	$350.00

PAGE 1

Figure 5-10. INVOICES Report Generated by INV-RPT Program

When Report Writer is used to generate a report, declarative procedures can be specified for execution during report generation. Each procedure is executed before a specific report group is written on the report file. If a SUPPRESS statement in the declarative procedure is executed, the report group is not written on the report file.

The sample program shown at the end of this section contains two declarative procedures. One of the declarative procedures includes the SUPPRESS statement.

SPECIFYING DECLARATIVES

Report writing declarative procedures are specified in the Declaratives portion of the Procedure Division. Each procedure is contained in a named section. The first statement in the section is a USE BEFORE REPORTING statement. This statement designates the data-name of the report group for which the procedure is to be executed.

The USE BEFORE REPORTING statement is followed by one or more paragraphs that are executed before the specified report group is written on the report file. The report writing declarative procedure includes Procedure Division statements that are executed each time the report group is processed. The following restrictions apply to report writing declarative procedures:

An INITIATE, GENERATE, or TERMINATE statement cannot be specified.

The value of a break control item for the report cannot be changed.

A group item that contains or is subordinate to a break control item cannot be referenced.

An item that redefines or renames any part of a break control item cannot be referenced.

SUPPRESSING THE PRINTING OF A REPORT GROUP

Within a declarative procedure, the printing of a report group is inhibited by execution of the SUPPRESS statement. This statement can only be specified in a USE FOR REPORTING declarative procedure. When the SUPPRESS statement is executed, all print lines in the report group named in the USE statement are suppressed. The LINE NUMBER and NEXT GROUP clauses contained in the report group are not processed; therefore, the special register LINE-COUNTER is not incremented.

When the line numbers specified for report groups overlap, the SUPPRESS statement should be used to suppress the printing of one of the report groups during execution. This can occur for a report heading and a page heading or for a page footing and a report footing. At the beginning of the report or at the end of the report, as applicable, one of the report groups must be suppressed. If the SUPPRESS statement is not used, an error exists and neither the heading nor the footing report group is printed.

Figure 6-1 shows the declarative procedures specified for the INVNTRY program. The first procedure, USE-PGFOOT SECTION, is executed each time a page footing is to be written on the report file. The data item END-FLAG is checked for a value of 1. When this condition is true, printing of the page footing is suppressed. If END-FLAG does not have a value of 1 the value of the data item NET2 is computed before the page heading is written on the report file. The computed value of NET2 is a source data item for the page footing.

```
PROCEDURE DIVISION.
DECLARATIVES.
USE-PGFOOT SECTION.
     USE BEFORE REPORTING PGFOOT.
USE-PG.
     IF END-FLAG EQUALS 1 SUPPRESS PRINTING
          ELSE COMPUTE NET2 = RTEMP - STEMP.
USE-RPFOOT SECTION.
     USE BEFORE REPORTING RPFOOT.
USE-RP.
     COMPUTE NET2 = RTEMP - STEMP.
END DECLARATIVES.
```

Figure 6-1. Declarative Procedures for INVNTRY Program

The second declarative procedure in figure 6-1, USE-RPFOOT SECTION, is executed only once. It is executed before the report footing report group is processed during report termination. The value of the data item NET2 is computed before the report footing report group is written on the report file.

INVNTRY PROGRAM

The report worksheet for the INVNTRY program is shown in figure 6-2; the program is shown in figure 6-3. This program reads an input file containing data that is used in the print lines and that is used to compute information to be printed. The two declarative procedures discussed in the preceding paragraphs are specified in the Declaratives portion of the Procedure Division (lines 70 through 80).

As each input record is read (line 88), the number of units received and the number of units shipped are added to two accumulators (lines 90 and 91). The number of units received minus the number of units shipped is then computed (line 92) to determine the net gain or net loss for the item, and the detail line is generated.

The page footing that is generated for each page includes accumulated totals for units shipped and units received. The net gain or net loss for all items up to the end of the page is computed when the declarative procedure (USE-PGFOOT SECTION) is executed (lines 71 through 75). When the end of the file is reached, the data item END-FLAG is set to one (line 97). Execution of the TERMINATE statement (line 98) causes the page footing report group to be processed; because END-FLAG is then equal to 1, printing of the page footing is suppressed on the last page. The report footing report group is processed and written on the report file after the declarative procedure (USE-RPFOOT SECTION) is executed (lines 76 through 79).

The input data illustrated in figure 6-4 is used to generate the report shown in figure 6-5.



Figure 6-2. Report Worksheet for INVNTRY Program

```
 1   IDENTIFICATION DIVISION.
 2   PROGRAM-ID. INVNTRY.
 3   ENVIRONMENT DIVISION.
 4   CONFIGURATION SECTION.
 5   SOURCE-COMPUTER.  CYBER-170.
 6   OBJECT-COMPUTER.  CYBER-170.
 7   INPUT-OUTPUT SECTION.
 8   FILE-CONTROL.
 9       SELECT IN-FILE ASSIGN TO ≠INPUT≠.
10       SELECT OUT-FILE   ASSIGN TO ≠OUTPUT≠.
11   DATA DIVISION.
12   FILE SECTION.
13   FD  IN-FILE
14           LABEL RECORD IS OMITTED
15           DATA RECORD IS IN-REC.
16   01  IN-REC.
17       03  STOCK-NO              PICTURE X(5).
18       03  FILLER                PICTURE XX.
19       03  UNITS-R               PICTURE 9(5).
20       03  FILLER                PICTURE XX.
21       03  UNITS-S               PICTURE 9(5).
22       03  FILLER                PICTURE X(61).
23   FD  OUT-FILE
24           LABEL RECORD IS OMITTED
25           REPORT IS INVENTORY.
26   WORKING-STORAGE SECTION.
27   77  RTEMP                     PICTURE 9(6).
28   77  STEMP                     PICTURE 9(6).
29   77  NET1                      PICTURE S9(5).
30   77  NET2                      PICTURE S9(7).
31   77  S-NO                      PICTURE X(5).
32   77  END-FLAG                  PICTURE 9       VALUE IS ZERO.
33   REPORT SECTION.
34   RD  INVENTORY
35           PAGE LIMIT IS 30 LINES
36               HEADING      3
37               FIRST DETAIL 8
38               LAST DETAIL 27.
39   01  TYPE IS REPORT HEADING  LINE NUMBER 3.
40       03  COLUMN NUMBER 53    PICTURE X(32)
41               VALUE IS ≠INVENTORY RECEIPTS AND SHIPMENTS≠.
42   01  TYPE IS PAGE HEADING    LINE NUMBER 6.
43       03  COLUMN NUMBER 26    PICTURE X(12)
44               VALUE IS ≠STOCK NUMBER≠.
45       03  COLUMN NUMBER 50    PICTURE X(14)
46               VALUE IS ≠UNITS RECEIVED≠.
47       03  COLUMN NUMBER 75    PICTURE X(13)
48               VALUE IS ≠UNITS SHIPPED≠.
49       03  COLUMN NUMBER 100   PICTURE X(13)
50               VALUE IS ≠NET GAIN/LOSS≠.
51   01  DET-LINE  TYPE IS DETAIL  LINE NUMBER PLUS 1.
52       03  COLUMN NUMBER 29    PICTURE X(5)     SOURCE IS STOCK-NO.
53       03  COLUMN NUMBER 54    PICTURE ZZ,ZZ9   SOURCE IS UNITS-R.
54       03  COLUMN NUMBER 79    PICTURE ZZ,ZZ9   SOURCE IS UNITS-S.
55       03  COLUMN NUMBER 103   PICTURE ---,--9 SOURCE IS NET1.
56   01  PGFOOT  TYPE IS PAGE FOOTING  LINE NUMBER 30.
57       03  COLUMN NUMBER 16    PICTURE X(23)
58               VALUE IS ≠ACCUMULATED TOTALS THRU≠.
59       03  COLUMN NUMBER 40    PICTURE X(5)       SOURCE IS S-NO.
60       03  COLUMN NUMBER 53    PICTURE ZZZ,ZZ9    SOURCE IS RTEMP.
61       03  COLUMN NUMBER 78    PICTURE ZZZ,ZZ9    SOURCE IS STEMP.
62       03  COLUMN NUMBER 102   PICTURE ----,--9   SOURCE IS NET2.
63   01  RPFOOT  TYPE IS REPORT FOOTING  LINE NUMBER 30.
64       03  COLUMN NUMBER 16    PICTURE X(26)
65               VALUE IS ≠TOTALS FOR ALL STOCK ITEMS≠.
66       03  COLUMN NUMBER 51    PICTURE Z,ZZZ,ZZ9    SOURCE IS RTEMP.
67       03  COLUMN NUMBER 76    PICTURE Z,ZZZ,ZZ9    SOURCE IS STEMP.
68       03  COLUMN NUMBER 100   PICTURE --,---,--9   SOURCE IS NET2.
69   PROCEDURE DIVISION.
70   DECLARATIVES.
71   USE-PGFOOT SECTION.
```

Figure 6-3. INVNTRY Program (Sheet 1 of 2)

```
    USE BEFORE REPORTING PGFOOT.
USE-PG.
    IF END-FLAG EQUALS 1 SUPPRESS PRINTING
        ELSE COMPUTE NET2 = RTEMP - STEMP.
USE-RPFOOT SECTION.
    USE BEFORE REPORTING RPFOOT.
USE-RP.
    COMPUTE NET2 = RTEMP - STEMP.
END DECLARATIVES.
REPORT-GEN SECTION.
STARTING.
    OPEN INPUT IN-FILE.
    OPEN OUTPUT OUT-FILE.
    INITIALIZE RTEMP, STEMP, NET1, NET2.
    INITIATE INVENTORY.
READ-IN.
    READ IN-FILE RECORD
        AT END GO TO FINISHED.
    ADD UNITS-R TO RTEMP.
    ADD UNITS-S TO STEMP.
    COMPUTE NET1 = UNITS-R - UNITS-S.
    GENERATE DET-LINE.
    MOVE STOCK-NO TO S-NO.
    GO TO READ-IN.
FINISHED.
    MOVE 1 TO END-FLAG.
    TERMINATE INVENTORY.
    CLOSE IN-FILE, OUT-FILE.
    STOP RUN.
```

Figure 6-3. INVNTRY Program (Sheet 2 of 2)

Column 1 Column 8 Column 15

```
AC149  01923  01146
AM802  24680  19235
BJ613  00475  01870
BP570  14336  14500
BZ294  36815  35232
CB321  08042  07051
CS988  12345  13254
DA004  05980  05365
DX823  21300  17848
EJ195  07461  09230
FR406  16270  18050
FZ618  33300  33000
GL117  00850  00695
GP530  01636  01544
GS066  04575  05216
HA924  12789  12458
JL285  16045  15775
JR303  09214  09930
KC850  04685  04526
LB147  08336  07941
MZ468  10210  10695
NG509  11688  11234
NP235  07205  06950
PA640  02936  02015
RB018  06314  06928
RT192  13456  12005
SA406  11823  10440
SE384  02695  02733
TJ951  09406  07155
TK820  12398  12490
VX265  04770  04288
WB473  08915  06512
WM502  01016  00801
WZ119  15267  13192
XK068  13478  14950
XR721  07104  06923
YB344  10980  10134
YN636  04322  06789
ZL870  06249  05006
```

Figure 6-4. Input Data for INVNTRY Program

INVENTORY RECEIPTS AND SHIPMENTS

STOCK NUMBER	UNITS RECEIVED	UNITS SHIPPED	NET GAIN/LOSS
AC149	1,923	1,146	777
AM802	24,680	19,235	5,445
BJ613	475	1,870	-1,395
BP570	14,336	14,500	-164
BZ294	36,815	35,232	1,583
CB321	8,042	7,051	991
CS988	12,345	13,254	-909
DA004	5,980	5,365	615
DX823	21,300	17,848	3,452
EJ195	7,461	9,230	-1,769
FR406	16,270	18,050	-1,780
FZ618	33,300	33,000	300
GL117	850	695	155
GP530	1,636	1,544	92
GS066	4,575	5,216	-641
HA924	12,789	12,458	331
JL285	16,045	15,775	270
JR303	9,214	9,930	-716
KC850	4,685	4,526	159
LB147	8,336	7,941	395
ACCUMULATED TOTALS THRU LB147	251,267	244,561	6,706

Figure 6-5. INVENTORY Report Generated by INVNTRY Program (Sheet 1 of 2)

STOCK NUMBER	UNITS RECEIVED	UNITS SHIPPED	NET GAIN/LOSS
MZ468	10,210	10,695	-485
NG509	11,688	11,234	454
NP235	7,205	6,950	255
PA640	2,936	2,015	921
RB018	6,314	6,928	-614
RT192	13,456	12,005	1,451
SA406	11,823	10,440	1,383
SE384	2,695	2,733	-38
TJ951	9,406	7,155	2,251
TK820	12,398	12,490	-92
VX265	4,770	4,288	482
WB473	8,915	6,512	2,403
WM502	1,016	801	215
WZ119	15,267	13,192	2,075
XK068	13,478	14,950	-1,472
XR721	7,104	6,923	181
YB344	10,980	10,134	846
YN636	4,322	6,789	-2,467
ZL870	6,249	5,006	1,243
TOTALS FOR ALL STOCK ITEMS	401,289	385,106	16,183

Figure 6-5. INVENTORY Report Generated by INVNTRY Program (Sheet 2 of 2)

STANDARD CHARACTER SETS A

CONTROL DATA operating systems offer the following variations of a basic character set:

CDC 64–character set

CDC 63–character set

ASCII 64–character set

ASCII 63–character set

The set in use at a particular installation was specified when the operating system was installed.

Depending on another installation option, the system assumes an input deck has been punched either in 026 or in 029 mode (regardless of the character set in use). Under NOS/BE 1, the alternate mode can be specified by a 26 or 29 punched in columns 79 and 80 of the job statement or any 7/8/9 card. The specified mode remains in effect through the end of the job unless it is reset by specification of the alternate mode on a subsequent 7/8/9 card.

Under NOS 1, the alternate mode can be specified by a 26 or 29 punched in columns 79 and 80 of any 6/7/9 card, as described above for a 7/8/9 card. In addition, 026 mode can be specified by a card with 5/7/9 multipunched in column 1, and 029 mode can be specified by a card with 5/7/9 multipunched in column 1 and a 9 punched in column 2.

Graphic character representation appearing at a terminal or printer depends on the installation character set and the terminal type. Characters shown in the CDC Graphic column of the standard character set table are applicable to BCD terminals; ASCII graphic characters are applicable to ASCII–CRT and ASCII–TTY terminals.

STANDARD CHARACTER SETS

CDC Graphic	ASCII Graphic Subset	Display Code	Hollerith Punch (026)	External BCD Code	ASCII Punch (029)	ASCII Code	CDC Graphic	ASCII Graphic Subset	Display Code	Hollerith Punch (026)	External BCD Code	ASCII Punch (029)	ASCII Code
:†	:	00††	8-2	00	8-2	072	6	6	41	6	06	6	066
A	A	01	12-1	61	12-1	101	7	7	42	7	07	7	067
B	B	02	12-2	62	12-2	102	8	8	43	8	10	8	070
C	C	03	12-3	63	12-3	103	9	9	44	9	11	9	071
D	D	04	12-4	64	12-4	104	+	+	45	12	60	12-8-6	053
E	E	05	12-5	65	12-5	105	-	-	46	11	40	11	055
F	F	06	12-6	66	12-6	106	*	*	47	11-8-4	54	11-8-4	052
G	G	07	12-7	67	12-7	107	/	/	50	0-1	21	0-1	057
H	H	10	12-8	70	12-8	110	(	(	51	0-8-4	34	12-8-5	050
I	I	11	12-9	71	12-9	111	)	)	52	12-8-4	74	11-8-5	051
J	J	12	11-1	41	11-1	112	$	$	53	11-8-3	53	11-8-3	044
K	K	13	11-2	42	11-2	113	=	=	54	8-3	13	8-6	075
L	L	14	11-3	43	11-3	114	blank	blank	55	no punch	20	no punch	040
M	M	15	11-4	44	11-4	115	, (comma)	, (comma)	56	0-8-3	33	0-8-3	054
N	N	16	11-5	45	11-5	116	. (period)	. (period)	57	12-8-3	73	12-8-3	056
O	O	17	11-6	46	11-6	117	≡	#	60	0-8-6	36	8-3	043
P	P	20	11-7	47	11-7	120	[	[	61	8-7	17	12-8-2	133
Q	Q	21	11-8	50	11-8	121	]	]	62	0-8-2	32	11-8-2	135
R	R	22	11-9	51	11-9	122	%	%	63††	8-6	16	0-8-4	045
S	S	23	0-2	22	0-2	123	≠	" (quote)	64	8-4	14	8-7	042
T	T	24	0-3	23	0-3	124	→	_ (underline)	65	0-8-5	35	0-8-5	137
U	U	25	0-4	24	0-4	125	∨	!	66	11-0 or 11-8-2†††	52	12-8-7 or 11-0†††	041
V	V	26	0-5	25	0-5	126							
W	W	27	0-6	26	0-6	127	∧	&	67	0-8-7	37	12	046
X	X	30	0-7	27	0-7	130	↑	' (apostrophe)	70	11-8-5	55	8-5	047
Y	Y	31	0-8	30	0-8	131	↓	?	71	11-8-6	56	0-8-7	077
Z	Z	32	0-9	31	0-9	132	<	<	72	12-0 or 12-8-2†††	72	12-8-4 or 12-0†††	074
0	0	33	0	12	0	060							
1	1	34	1	01	1	061	>	>	73	11-8-7	57	0-8-6	076
2	2	35	2	02	2	062	≤	@	74	8-5	15	8-4	100
3	3	36	3	03	3	063	≥	\	75	12-8-5	75	0-8-2	134
4	4	37	4	04	4	064	¬	^ (circumflex)	76	12-8-6	76	11-8-7	136
5	5	40	5	05	5	065	; (semicolon)	; (semicolon)	77	12-8-7	77	11-8-6	073

†Twelve or more zero bits at the end of a 60-bit word are interpreted as end-of-line mark rather than two colons. End-of-line mark is converted to external BCD 1632.

††In installations using a 63-graphic set, display code 00 has no associated graphic or card code; display code 63 is the colon (8-2 punch). The % graphic and related card codes do not exist and translations from ASCII/EBCDIC % yield a blank (55_8).

†††The alternate Hollerith (026) and ASCII (029) punches are accepted for input only.

CDC CHARACTER SET COLLATING SEQUENCE									
Collating Sequence Decimal/Octal		CDC Graphic	Display Code	External BCD	Collating Sequence Decimal/Octal		CDC Graphic	Display Code	External BCD
00	00	blank	55	20	32	40	H	10	70
01	01	≤	74	15	33	41	I	11	71
02	02	%	63 †	16 †	34	42	∨	66	52
03	03	[	61	17	35	43	J	12	41
04	04	→	65	35	36	44	K	13	42
05	05	≡	60	36	37	45	L	14	43
06	06	∧	67	37	38	46	M	15	44
07	07	↑	70	55	39	47	N	16	45
08	10	↓	71	56	40	50	O	17	46
09	11	>	73	57	41	51	P	20	47
10	12	≥	75	75	42	52	Q	21	50
11	13	¬	76	76	43	53	R	22	51
12	14	.	57	73	44	54	]	62	32
13	15	)	52	74	45	55	S	23	22
14	16	;	77	77	46	56	T	24	23
15	17	+	45	60	47	57	U	25	24
16	20	$	53	53	48	60	V	26	25
17	21	*	47	54	49	61	W	27	26
18	22	-	46	40	50	62	X	30	27
19	23	/	50	21	51	63	Y	31	30
20	24	,	56	33	52	64	Z	32	31
21	25	(	51	34	53	65	:	00 †	none†
22	26	=	54	13	54	66	0	33	12
23	27	≠	64	14	55	67	1	34	01
24	30	<	72	72	56	70	2	35	02
25	31	A	01	61	57	71	3	36	03
26	32	B	02	62	58	72	4	37	04
27	33	C	03	63	59	73	5	40	05
28	34	D	04	64	60	74	6	41	06
29	35	E	05	65	61	75	7	42	07
30	36	F	06	66	62	76	8	43	10
31	37	G	07	67	63	77	9	44	11

†In installations using the 63-graphic set, the % graphic does not exist. The : graphic is display code 63, External BCD code 16.

ASCII CHARACTER SET COLLATING SEQUENCE

Collating Sequence Decimal	Collating Sequence Octal	ASCII Graphic Subset	Display Code	ASCII Code	Collating Sequence Decimal	Collating Sequence Octal	ASCII Graphic Subset	Display Code	ASCII Code
00	00	blank	55	20	32	40	@	74	40
01	01	!	66	21	33	41	A	01	41
02	02	"	64	22	34	42	B	02	42
03	03	#	60	23	35	43	C	03	43
04	04	$	53	24	36	44	D	04	44
05	05	%	63†	25	37	45	E	05	45
06	06	&	67	26	38	46	F	06	46
07	07	'	70	27	39	47	G	07	47
08	10	(	51	28	40	50	H	10	48
09	11	)	52	29	41	51	I	11	49
10	12	*	47	2A	42	52	J	12	4A
11	13	+	45	2B	43	53	K	13	4B
12	14	,	56	2C	44	54	L	14	4C
13	15	-	46	2D	45	55	M	15	4D
14	16	.	57	2E	46	56	N	16	4E
15	17	/	50	2F	47	57	O	17	4F
16	20	0	33	30	48	60	P	20	50
17	21	1	34	31	49	61	Q	21	51
18	22	2	35	32	50	62	R	22	52
19	23	3	36	33	51	63	S	23	53
20	24	4	37	34	52	64	T	24	54
21	25	5	40	35	53	65	U	25	55
22	26	6	41	36	54	66	V	26	56
23	27	7	42	37	55	67	W	27	57
24	30	8	43	38	56	70	X	30	58
25	31	9	44	39	57	71	Y	31	59
26	32	:	00†	3A	58	72	Z	32	5A
27	33	;	77	3B	59	73	[	61	5B
28	34	<	72	3C	60	74	\	75	5C
29	35	=	54	3D	61	75	]	62	5D
30	36	>	73	3E	62	76	^	76	5E
31	37	?	71	3F	63	77	_	65	5F

†In installations using a 63-graphic set, the % graphic does not exist. The : graphic is display code 63.

64 CHARACTER EBCDIC SUBSET
COLLATING SEQUENCE

Collating Sequence Decimal/Octal		^Graphic	EBCDIC Punch	Display Code	EBCDIC Code
00	00	blank	no punch	55	40
01	01	.	12-8-3	57	4B
02	02	<	12-8-4	72	4C
03	03	(	12-8-5	51	4D
04	04	+	12-8-6	45	4E
05	05	|	12-8-7	66	4F
06	06	&	12	67	50
07	07	$	11-8-3	53	5B
08	10	*	11-8-4	47	5C
09	11	)	11-8-5	52	5D
10	12	;	11-8-6	77	5E
11	13	¬	11-8-7	76	5F
12	14	-	11	46	60
13	15	/	0-1	50	61
14	16	,	0-8-3	56	6B
15	17	%	0-8-4	63	6C
16	20	_	0-8-5	65	6D
17	21	>	0-8-6	73	6E
18	22	?	0-8-7	71	6F
19	23	:	8-2	00	7A
20	24	#	8-3	60	7B
21	25	@	8-4	74	7C
22	26	'	8-5	70	7D
23	27	=	8-6	54	7E
24	30	"	8-7	64	7F
25	31	¢	12-8-2/12-0	61	4A
26	32	A	12-1	01	C1
27	33	B	12-2	02	C2
28	34	C	12-3	03	C3
29	35	D	12-4	04	C4
30	36	E	12-5	05	C5
31	37	F	12-6	06	C6

64 CHARACTER EBCDIC SUBSET COLLATING SEQUENCE (Contd)				
Collating Sequence Decimal/Octal	Graphic	EBCDIC Punch	Display Code	EBCDIC Code
32 40	G	12-7	07	C7
33 41	H	12-8	10	C8
34 42	I	12-9	11	C9
35 43	!	11-8-2/11-0	62	5A
36 44	J	11-1	12	D1
37 45	K	11-2	13	D2
38 46	L	11-3	14	D3
39 47	M	11-4	15	D4
40 50	N	11-5	16	D5
41 51	O	11-6	17	D6
42 52	P	11-7	20	D7
43 53	Q	11-8	21	D8
44 54	R	11-9	22	D9
45 55	none	0-8-2	75	E0
46 56	S	0-2	23	E2
47 57	T	0-3	24	E3
48 60	U	0-4	25	E4
49 61	V	0-5	26	E5
50 62	W	0-6	27	E6
51 63	X	0-7	30	E7
52 64	Y	0-8	31	E8
53 65	Z	0-9	32	E9
54 66	0	0	33	F0
55 67	1	1	34	F1
56 70	2	2	35	F2
57 71	3	3	36	F3
58 72	4	4	37	F4
59 73	5	5	40	F5
60 74	6	6	41	F6
61 75	7	7	42	F7
62 76	8	8	43	F8
63 77	9	9	44	F9

UNIVAC 1108 COLLATING SEQUENCE [UNI]					
Collating Sequence Decimal/Octal		1108 Graphic	Card Punch	Display Code	CYBER Graphic
00	00	@	8-7	61	[
01	01	[	12-8-5	75	⩾
02	02	]	11-8-5	70	↑
03	03	⊬	12-8-7	77	;
04	04	Δ	11-8-7	73	>
05	05	blank	no punch	55	blank
06	06	A	12-1	01	A
07	07	B	12-1	02	B
08	10	C	12-3	03	C
09	11	D	12-4	04	D
10	12	E	12-5	05	E
11	13	F	12-6	06	F
12	14	G	12-7	07	G
13	15	H	12-8	10	H
14	16	I	12-9	11	I
15	17	J	11-1	12	J
16	20	K	11-2	13	K
17	21	L	11-3	14	L
18	22	M	11-4	15	M
19	23	N	11-5	16	N
20	24	O	11-6	17	O
21	25	P	11-7	20	P
22	26	Q	11-8	21	Q
23	27	R	11-9	22	R
24	30	S	0-2	23	S
25	31	T	0-3	24	T
26	32	U	0-4	25	U
27	33	V	0-5	26	V
28	34	W	0-6	27	W
29	35	X	0-7	30	X
30	36	Y	0-8	31	Y
31	37	Z	0-9	32	Z

UNIVAC 1108 COLLATING SEQUENCE [UNI] (Contd)					
Collating Sequence Decimal/Octal		1108 Graphic	Card Punch	Display Code	CYBER Graphic
32	40	)	12-8-4	52	)
33	41	–	11	46	–
34	42	+	12	45	+
35	43	<	12-8-6	76	¬
36	44	=	8-3	54	=
37	45	>	8-6	63	%
38	46	&	8-2	00	:
39	47	$	11-8-3	53	$
40	50	*	11-8-4	47	*
41	51	(	0-8-4	51	(
42	52	%	0-8-5	65	→
43	53	:	8-5	74	⩽
44	54	?	12-0	72	<
45	55	!	11-0	66	∨
46	56	,	0-8-3	56	,
47	58	\	0-8-6	60	≡
48	60	0	0	33	0̇
49	61	1	1	34	1
50	62	2	2	35	2
51	63	3	3	36	3
52	64	4	4	37	4
53	65	5	5	40	5
54	66	6	6	41	6
55	67	7	7	42	7
56	70	8	8	43	8
57	71	9	9	44	9
58	72	'	8-4	64	≠
59	73	;	11-8-6	71	↓
60	74	/	0-1	50	/
61	75	.	12-8-3	57	.
62	76	□	0-8-7	67	∧
63	77	≠	0-8-2	62	]

REPORT WRITER LANGUAGE SUMMARY B

A summary of the language formats applicable to Report Writer appears in this appendix. The page number listed for each format references the detailed information that can be found in the COBOL 5 Reference Manual. The following elements are alphabetized in one list:

RD entry

FD entry

Report group description entry

Data Division clauses, by clause name

Procedure Division statements, by statement name

Page

INITIATE Statement

6-11

```
INITIATE report-name-1 [, report-name-2] . . .
```

JUSTIFIED Clause

4-10

```
{JUSTIFIED} RIGHT
{JUST     }
```

LINE NUMBER Clause

6-6

```
LINE NUMBER IS {integer-1 [ON NEXT PAGE]}
               {PLUS integer-2           }
```

NEXT GROUP Clause

6-7

```
NEXT GROUP IS {integer-1      }
              {PLUS integer-2 }
              {NEXT PAGE      }
```

PAGE Clause

6-3

```
PAGE [LIMIT IS ] integer-1 [LINE ] [, HEADING integer-2]
     [LIMITS ARE]          [LINES]

    [, FIRST DETAIL integer-3] [, LAST DETAIL integer-4]

    [, FOOTING integer-5]
```

PICTURE Clause

4-12

```
{PICTURE} IS character-string
{PIC    }
```

RD Entry in Report Section

6-2

```
RD report-name
    [; CODE clause]
    [; CONTROL clause]
    [; PAGE clause].
    {report-group-description entry} . . .
```

REPORT Clause

6-2

```
{REPORT IS } report-name-1 [, report-name-2] . . .
{REPORTS ARE}
```

Page

Report Group Description Entry Format 1 — 6-5

```
01  [data-name]
    [; LINE NUMBER clause]
    [; NEXT GROUP clause]
    ; TYPE clause
    [; USAGE clause] .
```

Report Group Description Entry Format 2 — 6-5

```
level-number  [data-name]
    [; LINE NUMBER clause]
    [; USAGE clause] .
```

Report Group Description Entry Format 3 — 6-5

```
level-number  [data-name]
    [; BLANK WHEN ZERO clause]
    [; COLUMN NUMBER clause]
    [; GROUP INDICATE clause]
    [; JUSTIFIED clause]
    [; LINE NUMBER clause]
    ; PICTURE clause
    { ; SOURCE clause }
    { ; SUM clause    }
    { ; VALUE clause  }
    [; USAGE clause] .
```

Report Section of Data Division — 6-2

```
REPORT SECTION.
{RD entry}
[report group description entry] . . .
```

SOURCE Clause — 6-7

```
SOURCE IS identifier
```

SUM Clause — 6-7

```
SUM identifier-1 [, identifier-2] . . . [UPON data-name-1 [, data-name-2] . . .] . . .
    [RESET ON {data-name-3}]
              {FINAL      }
```

Page

SUPPRESS Statement 6-11

SUPPRESS PRINTING

TERMINATE Statement 6-11

TERMINATE report-name-1 [, report-name-2] . . .

TYPE Clause 6-9



USAGE Clause 6-10

USAGE IS DISPLAY

USE Statement Format 2 6-12

USE BEFORE REPORTING identifier.

VALUE Clause Format 1 4-19

VALUE IS literal

REPORT GENERATION WITHOUT REPORT WRITER C

This appendix contains a COBOL 5 program that generates a report without using the Report Writer feature. For comparison purposes, the report produced by this program is the same report that is generated by the BUDGETS program shown in figure 4-12.

Figure C-1 shows the BUDGETS program without Report Writer. The input data shown in figure C-2 is used to print the report shown in figure C-3.

Each type of line to be printed on the report is described in the Working-Storage Section (lines 33 through 129). The eight sum counters, a page counter, and a data item for the prior value of ACCOUNT are also defined in the Working-Storage Section (lines 131 through 140).

Before the first detail line is written, the report is initialized. The sum counters are set to zero (lines 146 and 147). The report heading, page heading, final heading, and control group headings are written on the report (lines 148 through 154).

As each input record is read, the DEPT, SECT, and ACCOUNT data items are checked for a change in value (lines 159 through 169). If a value changes, the applicable control footing lines and control heading lines (lines 178 through 222) are written on the report before the detail line is printed (lines 173 and 174). The eight sum counters are then incremented by the values in the input record (lines 175 and 176). Resetting the sum counters to zero occurs after the applicable control footing has been written (lines 189, 196, and 200).

At report termination, the control footings for the last group of detail lines and then the grand totals and the page footing are written on the report (lines 227 through 234).

```
 1   IDENTIFICATION DIVISION.
 2   PROGRAM-ID.  BUDGETS.
 3   ENVIRONMENT DIVISION.
 4   CONFIGURATION SECTION.
 5   SOURCE-COMPUTER.  CYBER-170.
 6   OBJECT-COMPUTER.  CYBER-170.
 7   INPUT-OUTPUT SECTION.
 8   FILE-CONTROL.
 9       SELECT CARD-IN ASSIGN TO ≠INPUT≠.
10       SELECT PRINT-FILE ASSIGN TO ≠OUTPUT≠.
11   DATA DIVISION.
12   FILE SECTION.
13   FD  CARD-IN
14           LABEL RECORDS ARE OMITTED
15           DATA RECORD IS CARD-REC.
16   01  CARD-REC.
17       03  DEPT                PICTURE X(5).
18       03  FILLER              PICTURE XXX.
19       03  SECT                PICTURE X(4).
20       03  FILLER              PICTURE XXX.
21       03  ACCOUNT             PICTURE 9(5).
22       03  FILLER              PICTURE XXX.
23       03  BUDGET              PICTURE 9(4)V99.
24       03  FILLER              PICTURE XXX.
25       03  ACTUAL              PICTURE 9(4)V99.
26       03  FILLER              PICTURE X(42).
27   FD  PRINT-FILE
28           LABEL RECORDS ARE OMITTED
29           LINAGE IS 55 LINES
30           DATA RECORD IS PRINT-LINE.
31   01  PRINT-LINE              PICTURE X(136).
32   WORKING-STORAGE SECTION.
33   01  REPORT-HEAD.
34       03  FILLER              PICTURE X(42)   VALUE IS SPACES.
35       03  FILLER              PICTURE X(52)   VALUE IS ≠B U D G E T
36   -           ≠V S    A C T U A L    E X P E N S E S≠.
37       03  FILLER              PICTURE X(42)   VALUE IS SPACES.
38   01  PAGE-HEAD1.
39       03  FILLER              PICTURE X(60)   VALUE IS SPACES.
40       03  FILLER              PICTURE X(6)    VALUE IS ≠BUDGET≠.
41       03  FILLER              PICTURE X(34)   VALUE IS SPACES.
42       03  FILLER              PICTURE X(6)    VALUE IS ≠ACTUAL≠.
43       03  FILLER              PICTURE X(30)   VALUE IS SPACES.
44   01  PAGE-HEAD2.
45       03  FILLER              PICTURE X(48)   VALUE IS SPACES.
46       03  FILLER              PICTURE X(30)   VALUE IS ALL ≠-≠.
47       03  FILLER              PICTURE X(10)   VALUE IS SPACES.
48       03  FILLER              PICTURE X(30)   VALUE IS ALL ≠-≠.
49       03  FILLER              PICTURE X(18)   VALUE IS SPACES.
50   01  FINAL-HEAD.
51       03  FILLER              PICTURE X(18)   VALUE IS SPACES.
52       03  FILLER              PICTURE X(17)   VALUE ≠***REPORT COVERS≠.
53       03  MONTH-IN            PICTURE XX.
54       03  FILLER              PICTURE X(10)   VALUE IS ≠ MONTHS***≠.
55       03  FILLER              PICTURE X(89)   VALUE IS SPACES.
56   01  DEPT-HEAD.
57       03  FILLER              PICTURE X(21)   VALUE IS SPACES.
58       03  FILLER              PICTURE X(11)   VALUE IS ≠DEPARTMENT≠.
59       03  DEPT-OUT            PICTURE X(5).
60       03  FILLER              PICTURE X(99)   VALUE IS SPACES.
61   01  SECT-HEAD.
62       03  FILLER              PICTURE X(24)   VALUE IS SPACES.
63       03  FILLER              PICTURE X(8)    VALUE IS ≠SECTION≠.
64       03  SECT-OUT            PICTURE X(4).
65       03  FILLER              PICTURE X(100)  VALUE IS SPACES.
66   01  ACCT-HEAD.
67       03  FILLER              PICTURE X(32)   VALUE IS SPACES.
68       03  FILLER              PICTURE X(7)    VALUE IS ≠ACCOUNT≠.
69       03  FILLER              PICTURE X(97)   VALUE IS SPACES.
70   01  DETAIL-LINE.
```

Figure C-1. BUDGETS Program Without the Report Writer Feature (Sheet 1 of 4)

```
 71        03  FILLER              PICTURE X(33)   VALUE IS SPACES.
 72        03  ACCOUNT-OUT        PICTURE X(5).
 73        03  FILLER              PICTURE X(11)   VALUE IS SPACES.
 74        03  BUDGET-OUT         PICTURE $$,999.99.
 75        03  FILLER              PICTURE X(31)   VALUE IS SPACES.
 76        03  ACTUAL-OUT         PICTURE $$,999.99.
 77        03  FILLER              PICTURE X(38)   VALUE IS SPACES.
 78    01  ACCT-FOOT1.
 79        03  FILLER              PICTURE X(48)   VALUE IS SPACES.
 80        03  FILLER              PICTURE X(10)   VALUE IS ALL ≠-≠.
 81        03  FILLER              PICTURE X(30)   VALUE IS SPACES.
 82        03  FILLER              PICTURE X(10)   VALUE IS ALL ≠-≠.
 83        03  FILLER              PICTURE X(38)   VALUE IS SPACES.
 84    01  ACCT-FOOT2.
 85        03  FILLER              PICTURE X(48)   VALUE IS SPACES.
 86        03  BUD-1-OUT           PICTURE $$$,999.99.
 87        03  FILLER              PICTURE X(30)   VALUE IS SPACES.
 88        03  ACT-1-OUT           PICTURE $$$,999.99.
 89        03  FILLER              PICTURE X(38)   VALUE IS SPACES.
 90    01  ACCT-FOOT3.
 91        03  FILLER              PICTURE X(24)   VALUE IS SPACES.
 92        03  FILLER              PICTURE X(22)   VALUE IS ≠ACCUM TOTALS -
 93    -       ≠SECTION≠.
 94        03  FILLER              PICTURE X(21)   VALUE IS SPACES.
 95        03  BUD-2-OUT           PICTURE $$$$,999.99.
 96        03  FILLER              PICTURE X(29)   VALUE IS SPACES.
 97        03  ACT-2-OUT           PICTURE $$$$,999.99.
 98        03  FILLER              PICTURE X(18)   VALUE IS SPACES.
 99    01  SECT-FOOT.
100        03  FILLER              PICTURE X(21)   VALUE IS SPACES.
101        03  FILLER              PICTURE X(25)   VALUE IS ≠ACCUM TOTALS -
102    -       ≠DEPARTMENT≠.
103        03  FILLER              PICTURE X(19)   VALUE IS SPACES.
104        03  BUD-3-OUT           PICTURE $$,$$$,999.99.
105        03  FILLER              PICTURE X(27)   VALUE IS SPACES.
106        03  ACT-3-OUT           PICTURE $$,$$$,999.99.
107        03  FILLER              PICTURE X(18)   VALUE IS SPACES.
108    01  DEPT-FOOT.
109        03  FILLER              PICTURE X(18)   VALUE IS SPACES.
110        03  FILLER              PICTURE X(100)  VALUE IS ALL ≠*≠.
111        03  FILLER              PICTURE X(18)   VALUE IS SPACES.
112    01  FINAL-FOOT1.
113        03  FILLER              PICTURE X(15)   VALUE IS SPACES.
114        03  FILLER              PICTURE X(106)  VALUE IS ALL ≠*≠.
115        03  FILLER              PICTURE X(15)   VALUE IS SPACES.
116    01  FINAL-FOOT2.
117        03  FILLER              PICTURE X(18)   VALUE IS SPACES.
118        03  FILLER              PICTURE X(30)   VALUE IS ≠GRAND TOTALS -
119    -       ≠ALL DEPARTMENTS≠.
120        03  FILLER              PICTURE X(16)   VALUE IS SPACES.
121        03  BUD-4-OUT           PICTURE $$$,$$$,999.99.
122        03  FILLER              PICTURE X(26)   VALUE IS SPACES.
123        03  ACT-4-OUT           PICTURE $$$,$$$,999.99.
124        03  FILLER              PICTURE X(18)   VALUE IS SPACES.
125    01  PAGE-FOOT.
126        03  FILLER              PICTURE X(113)  VALUE IS SPACES.
127        03  FILLER              PICTURE X(5)    VALUE IS ≠PAGE ≠.
128        03  PAGE-OUT            PICTURE ZZ9.
129        03  FILLER              PICTURE X(15)   VALUE IS SPACES.
130    77  TEMP                    PICTURE 99.
131    77  PAGE-NO                 PICTURE 999     VALUE IS ZERO.
132    77  BUD-1                   PICTURE 9(5)V99.
133    77  BUD-2                   PICTURE 9(6)V99.
134    77  BUD-3                   PICTURE 9(7)V99.
135    77  BUD-4                   PICTURE 9(8)V99.
136    77  ACT-1                   PICTURE 9(5)V99.
137    77  ACT-2                   PICTURE 9(6)V99.
138    77  ACT-3                   PICTURE 9(7)V99.
139    77  ACT-4                   PICTURE 9(8)V99.
140    77  ACCT                    PICTURE 9(5).
```

Figure C-1. BUDGETS Program Without the Report Writer Feature (Sheet 2 of 4)

```
141   PROCEDURE DIVISION.
142   INITIALIZATION.
143       OPEN INPUT CARD-IN.
144       OPEN OUTPUT PRINT-FILE.
145       ACCEPT MONTH-IN.
146       INITIALIZE BUD-1, BUD-2, BUD-3, BUD-4,
147           ACT-1, ACT-2, ACT-3, ACT-4.
148       WRITE PRINT-LINE FROM REPORT-HEAD
149           AFTER ADVANCING 1 LINE.
150       PERFORM HEADINGS.
151       WRITE PRINT-LINE FROM FINAL-HEAD.
152       READ CARD-IN RECORD
153           AT END GO TO ERR-1.
154       PERFORM DEPT-BREAK-HEAD THRU ACCT-BREAK-HEAD.
155       GO TO PRINT-DETAIL.
156   READ-CARD.
157       READ CARD-IN RECORD
158           AT END GO TO CLOSING.
159       IF DEPT NOT EQUAL TO DEPT-OUT
160           PERFORM ACCT-BREAK-FOOT THRU ACCT-BREAK-HEAD
161           GO TO PRINT-DETAIL.
162       IF SECT NOT EQUAL TO SECT-OUT
163           PERFORM ACCT-BREAK-FOOT THRU SECT-BREAK-FOOT
164           PERFORM SECT-BREAK-HEAD THRU ACCT-BREAK-HEAD
165           GO TO PRINT-DETAIL.
166       IF ACCOUNT NOT EQUAL TO ACCT
167           PERFORM ACCT-BREAK-FOOT
168           PERFORM ACCT-BREAK-HEAD
169           ELSE MOVE SPACES TO ACCOUNT-OUT.
170   PRINT-DETAIL.
171       MOVE BUDGET TO BUDGET-OUT.
172       MOVE ACTUAL TO ACTUAL-OUT.
173       WRITE PRINT-LINE FROM DETAIL-LINE
174           AT EOP PERFORM PAGE-FOOTING THRU HEADINGS.
175       ADD BUDGET TO BUD-1, BUD-2, BUD-3, BUD-4.
176       ADD ACTUAL TO ACT-1, ACT-2, ACT-3, ACT-4.
177       GO TO READ-CARD.
178   ACCT-BREAK-FOOT.
179       WRITE PRINT-LINE FROM ACCT-FOOT1
180           AT EOP PERFORM PAGE-FOOTING THRU HEADINGS.
181       MOVE BUD-1 TO BUD-1-OUT.
182       MOVE ACT-1 TO ACT-1-OUT.
183       WRITE PRINT-LINE FROM ACCT-FOOT2
184           AT EOP PERFORM PAGE-FOOTING THRU HEADINGS.
185       MOVE BUD-2 TO BUD-2-OUT.
186       MOVE ACT-2 TO ACT-2-OUT.
187       WRITE PRINT-LINE FROM ACCT-FOOT3
188           AT EOP PERFORM PAGE-FOOTING THRU ACCT-BREAK-HEAD.
189       INITIALIZE BUD-1, ACT-1.
190   SECT-BREAK-FOOT.
191       MOVE BUD-3 TO BUD-3-OUT.
192       MOVE ACT-3 TO ACT-3-OUT.
193       WRITE PRINT-LINE FROM SECT-FOOT
194           AFTER ADVANCING 2 LINES
195           AT EOP PERFORM PAGE-FOOTING THRU ACCT-BREAK-HEAD.
196       INITIALIZE BUD-2, ACT-2.
197   DEPT-BREAK-FOOT.
198       WRITE PRINT-LINE FROM DEPT-FOOT
199           AFTER ADVANCING 2 LINES.
200       INITIALIZE BUD-3, ACT-3.
201   PAGE-FOOTING.
202       SUBTRACT LINAGE-COUNTER FROM 55 GIVING TEMP.
203       ADD 1 TO PAGE-NO.
204       MOVE PAGE-NO TO PAGE-OUT.
205       WRITE PRINT-LINE FROM PAGE-FOOT
206           AFTER ADVANCING TEMP LINES.
207   HEADINGS.
208       WRITE PRINT-LINE FROM PAGE-HEAD1
209           AFTER ADVANCING 3 LINES.
210       WRITE PRINT-LINE FROM PAGE-HEAD2.
```

Figure C-1. BUDGETS Program Without the Report Writer Feature (Sheet 3 of 4)

```
211    DEPT-BREAK-HEAD.
212        MOVE DEPT TO DEPT-OUT.
213        WRITE PRINT-LINE FROM DEPT-HEAD
214            AFTER ADVANCING 3 LINES.
215    SECT-BREAK-HEAD.
216        MOVE SECT TO SECT-OUT.
217        WRITE PRINT-LINE FROM SECT-HEAD
218            AFTER ADVANCING 2 LINES.
219    ACCT-BREAK-HEAD.
220        MOVE ACCOUNT TO ACCOUNT-OUT, ACCT.
221        WRITE PRINT-LINE FROM ACCT-HEAD
222            AFTER ADVANCING 2 LINES.
223    ERR-1.
224        DISPLAY #NO INPUT RECORDS#
225        STOP RUN.
226    CLOSING.
227        PERFORM ACCT-BREAK-FOOT THRU DEPT-BREAK-FOOT.
228        MOVE BUD-4 TO BUD-4-OUT.
229        MOVE ACT-4 TO ACT-4-OUT.
230        WRITE PRINT-LINE FROM FINAL-FOOT1
231            AFTER ADVANCING 3 LINES.
232        WRITE PRINT-LINE FROM FINAL-FOOT2.
233        WRITE PRINT-LINE FROM FINAL-FOOT1.
234        PERFORM PAGE-FOOTING.
235        CLOSE CARD-IN, PRINT-FILE.
236        STOP RUN.
```

Figure C-1. BUDGETS Program Without the Report Writer Feature (Sheet 4 of 4)

```
                6
Column 1    Column 9        Column 16  Column 24    Column 33
        PRSNL   PS10    24689   102500   114650
        PRSNL   PS10    24689   085000   067500
        PRSNL   PS10    24689   243800   238160
        PRSNL   PS10    48153   505000   519275
        PRSNL   PS30    24689   116500   094150
        PRSNL   PS30    24689   305000   332525
        PRSNL   PS30    48153   813500   806975
        PRGMG   PG25    51960   648500   612800
        PRGMG   PG25    51960   284000   356045
        PRGMG   PG25    87013   056500   041580
        PRGMG   PG25    87013   124900   116250
        PRGMG   PG80    51960   450000   518575
        PRGMG   PG80    87013   739500   772125
        PRGMG   PG80    87013   541600   581050
        ACCTG   AC45    10495   362500   332190
        ACCTG   AC45    62377   601600   615570
        ACCTG   AC45    62377   495400   493200
        ACCTG   AC45    62377   326500   331585
        ACCTG   AC70    10495   518600   452910
        ACCTG   AC70    10495   247500   215625
        ACCTG   AC70    62377   664000   653550
```

Figure C-2. Input Data for the Report

```
                              B U D G E T   V S   A C T U A L   E X P E N S E S

                                                    BUDGET                                ACTUAL
                                          ------------------------------    ------------------------------
***REPORT COVERS  6 MONTHS***

   DEPARTMENT PRSNL

      SECTION PS10

              ACCOUNT
               24689                      $1,025.00                          $1,146.50
                                            $850.00                            $675.00
                                          $2,438.00                          $2,381.60
                                          ---------                          ---------
                                          $4,313.00                          $4,203.10
      ACCUM TOTALS - SECTION                              $4,313.00                             $4,203.10

              ACCOUNT
               48153                      $5,050.00                          $5,192.75
                                          ---------                          ---------
                                          $5,050.00                          $5,192.75
      ACCUM TOTALS - SECTION                              $9,363.00                             $9,395.85

   ACCUM TOTALS - DEPARTMENT                              $9,363.00                             $9,395.85

      SECTION PS30

              ACCOUNT
               24689                      $1,165.00                            $941.50
                                          $3,050.00                          $3,325.25
                                          ---------                          ---------
                                          $4,215.00                          $4,266.75
      ACCUM TOTALS - SECTION                              $4,215.00                             $4,266.75

              ACCOUNT
               48153                      $8,135.00                          $8,069.75
                                          ---------                          ---------
                                          $8,135.00                          $8,069.75
      ACCUM TOTALS - SECTION                             $12,350.00                            $12,336.50

   ACCUM TOTALS - DEPARTMENT                             $21,713.00                            $21,732.35

**********************************************************************************************************


                                                                                                PAGE    1
```

Figure C-3. Report Output by the Program (Sheet 1 of 3)

	BUDGET		ACTUAL	
DEPARTMENT PRGMG				
SECTION PG25				
ACCOUNT 51960	$6,485.00		$6,128.00	
	$2,840.00		$3,560.45	
	$9,325.00		$9,688.45	
ACCUM TOTALS - SECTION		$9,325.00		$9,688.45
ACCOUNT 87013	$565.00		$415.80	
	$1,249.00		$1,162.50	
	$1,814.00		$1,578.30	
ACCUM TOTALS - SECTION		$11,139.00		$11,266.75
ACCUM TOTALS - DEPARTMENT		$11,139.00		$11,266.75
SECTION PG80				
ACCOUNT 51960	$4,500.00		$5,185.75	
	$4,500.00		$5,185.75	
ACCUM TOTALS - SECTION		$4,500.00		$5,185.75
ACCOUNT 87013	$7,395.00		$7,721.25	
	$5,416.00		$5,810.50	
	$12,811.00		$13,531.75	
ACCUM TOTALS - SECTION		$17,311.00		$18,717.50
ACCUM TOTALS - DEPARTMENT		$28,450.00		$29,984.25

****

PAGE 2

Figure C-3. Report Output by the Program (Sheet 2 of 3)

```
                                        BUDGET                          ACTUAL
                                -----------------------         -----------------------

DEPARTMENT ACCTG

   SECTION AC45

           ACCOUNT
            10495               $3,625.00                       $3,321.90
                                ---------                       ---------
                                $3,625.00                       $3,321.90
   ACCUM TOTALS - SECTION                       $3,625.00                       $3,321.90

           ACCOUNT
            62377               $6,016.00                       $6,155.70
                                $4,954.00                       $4,932.00
                                $3,265.00                       $3,315.85
                                ---------                       ---------
                                $14,235.00                      $14,403.55
   ACCUM TOTALS - SECTION                       $17,860.00                      $17,725.45

ACCUM TOTALS - DEPARTMENT                       $17,860.00                      $17,725.45

   SECTION AC70

           ACCOUNT
            10495               $5,186.00                       $4,529.10
                                $2,475.00                       $2,156.25
                                ---------                       ---------
                                $7,661.00                       $6,685.35
   ACCUM TOTALS - SECTION                       $7,661.00                       $6,685.35

           ACCOUNT
            62377               $6,640.00                       $6,535.50
                                ---------                       ---------
                                $6,640.00                       $6,535.50
   ACCUM TOTALS - SECTION                       $14,301.00                      $13,220.85

ACCUM TOTALS - DEPARTMENT                       $32,161.00                      $30,946.30

**********************************************************************************************************

**********************************************************************************************************
   GRAND TOTALS - ALL DEPARTMENTS               $82,324.00                      $82,662.90
**********************************************************************************************************

                                                                                        PAGE    3
```

Figure C-3. Report Output by the Program (Sheet 3 of 3)

INDEX

COMMENT SHEET

MANUAL TITLE: COBOL Version 5 REPORT WRITER User Guide

PUBLICATION NO.: 60496900 **REVISION:** A

NAME: _____

COMPANY: _____

STREET ADDRESS: _____

CITY: _____ **STATE:** _____ **ZIP CODE:** _____

This form is not intended to be used as an order blank. Control Data Corporation welcomes your evaluation of this manual. Please indicate any errors, suggested additions or deletions, or general comments below (please include page number references).

☐ Please reply ☐ No reply necessary

NO POSTAGE STAMP NECESSARY IF MAILED IN U.S.A.

FOLD ON DOTTED LINES AND TAPE

AA3419 REV. 4/79 PRINTED IN U.S.A.

TAPE

TAPE

FOLD

FOLD

NO POSTAGE
NECESSARY
IF MAILED
IN THE
UNITED STATES

BUSINESS REPLY MAIL

FIRST CLASS PERMIT NO. 8241 MINNEAPOLIS, MINN.

POSTAGE WILL BE PAID BY

CONTROL DATA CORPORATION

Publications and Graphics Division

215 Moffett Park Drive
Sunnyvale, California 94086

CUT ALONG LINE

FOLD

FOLD

CORPORATE HEADQUARTERS, P.O. BOX 0, MINNEAPOLIS, MINN. 55440
SALES OFFICES AND SERVICE CENTERS IN MAJOR CITIES THROUGHOUT THE WORLD

LITHO IN U.S

CONTROL DATA CORPORATION